

ファミコンを  
どんどん使いこなせ

ファミコン大作戦

# ファミコン、ステッパ

TM

もうきみは  
ベータシッター  
ファイターだ!

地球防衛軍 編著  
司令 谷岡康則



倉敷市立中央図書館



2010217012

誠文堂新光社



## ファミコン大作戦

ファミコンを使いこなすための第1弾

### ファミリーベーシック ホップ

キーボードの打ち込み方から、ゲーム  
プログラム完成まで。

君のファミコンでチャレンジだ！


☆作戦指令001 キーボードを使いこなそう

☆作戦指令002 BASICでの計算は？

☆作戦指令003 プログラムに挑戦だ！

地球防衛軍 編著 B5変形108頁 定価800円







ニ

キ

地

フ


き



スエミツン大作戦

スエミツンパーキング

ステップ





# HOP STEP JUMP

ホップ編を読んでくれた君、ありがとう。  
そして、ステップ編から参加してくれる君、がんばろうね！  
ステップ編では、ベーシックについて少しつっこんでみよう。  
今までのいろんなコンピューターの入門書って、こうしたらこうなります、こんなことは  
しちゃだめだよ、としか書いてなかったよね。  
でも、なぜ、こんなことをするとエラーになるのか？  
なぜ、こんな命令が必要なのか？  
こんな疑問には、答えてくれなかっただろう。  
この本も、ホップ編のように、最初のページから順番に読んでほしいんだ。そして、書かれ  
てある短いプログラムを自分で打ち込んでみて、プログラムの数値をいろいろ変えてみても  
らいたいんだ。  
プログラムには、「たった一つの正解」というのはないから、あることをするには、いろい  
ろな方法が考えられるんだ。大事なものはその考え方なんだよ。  
この本では、単なるベーシックの解説じゃなく、ここをこうしたらどうなるんだろう？こ  
うやったらいいんじゃないか？といろんなやり方を考えてみたんだ。



ファミコンだけでなく、他のパソコンはどうなんだろう？というような疑問にも、ふれて  
いるんだよ。

もしかしたら、ファミコンだけで遊ぶには、必要のないこともあるかも知れないけど、そ  
れを知ること、君はまた一歩、コンピューターについての理解が深まるんだ、と信じてい  
るよ。

教えられたこと、書いてあることなら知ってるけど、あとは知らない、それじゃあ、つま  
らないと思わないかい？

コンピューターと人間の一番大きな違いは、


「なぜ？」

と疑問を持てるか、持てないか、ということではないのかな？

いっしょに考えよう。考えながらこのステップ編をマスターした君は、もう友達にも教え  
ることができるようになるはずぞ！

君も、そして、君のお父さん、お母さん、お兄さんにも一諸に参加してほしい、と思っ  
ています。

1985年4月20日 谷岡 康則





# もくじ

## FAMILY STEP BASIC

### 作戦指令004

#### 少しうしろをふり返ってみよう

004-1 ダイレクト命令はどれぐらい? ⑦

004-2 たった8個の命令で ⑭

004-3 コンピューターに入力するのは? ⑲

### 作戦指令005

#### ベーシック自由自在

005-1 あっちの道、こっちの道 ⑳

005-2 くり返して何回も ㉔

005-3 文字と数字のあいだから ㉘

●制作・企画 地球防衛軍 谷岡康則・岩田 裕・鈴木康夫・真藤美紀子・

簗藤 桂・北村 順・初瀬 籠・美崎靖之

●編集・企画 リポート社



## 作戦指令006

### テレビへ落書き、3つの方法

- 006-1 PRINT文にも、いろいろあるぞ! ⑤6
- 006-2 ファミコンは何重人格? ⑥1
- 006-3 スプライトが重なると。⑥7
- 006-4 スプライトを動かすには? ⑦3
- 006-5 MOVEというのは、動くという意味 ⑦7

## 作戦指令007

### おっかけゲームに挑戦だ!

- 007-1 どんなゲームを作ろうか? ⑧6
- 007-2 どっちの方向なら動けるだろう? ⑧9
- 007-3 敵はおりこうさん? ⑨2
- 007-4 最後の仕上げはこうやって ⑨4

#### 巻末資料

- 作戦資料101 緊急情報
- 作戦資料102 ROMとRAM
- 作戦資料103 コントロール・キーの使い方
- キャラクターテーブル(A) ●キャラクターテーブル(B)

●カバー・本文イラスト 高橋裕之 (ファクトリー)

© 1985 Seibundo Shinkosha Publishing Co.,Ltd. ※本誌に掲載の記事は、無断転用を禁じます。



作戦指令  
004

少しうしろを  
ふり返ってみよう









1 Ø A=1:PRINT A

2 Ø RUN

と入力してみよう。「RUN」のあとで、リターン・キーを打ち込んでも何もおきないね。つまり、ダイレクトに「RUN」と入力しない限り、コンピューターは「プログラム」を実行しないんだ。もちろん、10行の「PRINT A」という命令も実行しないことになるね。それじゃ、このプログラムを実行するとどうなるかな？

RUN

と入力してみよう。画面には、たてに「1」という数字が次々に出てくるね。ストップ・キーを押さない限り、止まらない。

「RUN」とダイレクトに入力したのは、1回だけなのに、コンピューターは何回もプログラムをくり返し実行しているみたいだ？


そう、その通りなんだ。「RUN」と命令されたコンピューターは、10行の命令を実行し、画面に「1」と表示して、次の20行へいく。20行では「RUN」と書かれているプログラムを実行する、ということとは、順番にその行に書かれている命令を実行するわけだから、「RUN」という命令を実行して、また、10行へ行くんだ。

ストップ・キーでむりやり実行を止めるまでは、この動作を永遠にくり返し続けているわけだ。

少し頭がこんがらがってきたかな？ダイレクト・モードでは、一つの命令を実行するには、リターン・キーを押す必要があったね。リターン・キーが命令をコンピューターまで届ける役割をしていたんだ。そして、この「RUN」命令は、プログラムの一つ一つの命令を自動的にコンピューターに送り込む、ベルトコンベアーみたいなもの、と思えばいいね。

ところで、この「RUN」という命令には、もう一つ働きがあるんだ。その前に、このプログラムを一度消してみよう。

NEW



NEW

とダイレクトに入力してみる。そのあと「LIST」ととっても何も表示されないね。そう、「NEW」はプログラムを全部消す命令だったね。

さて、

A=1……①



```
OK  
NEW  
OK  
LIST  
OK  
■
```

```
OK  
A=1  
OK  
PRINT A  
1  
OK  
10 PRINT A.  
PRINT A  
0  
OK  
A=1:RUN  
0  
OK  
■
```



とダイレクトで打ち込んでほしい。次に、

**PRINT A.....②**

と入力してみると、ちゃんと「 1」と表示されるね。

**10 PRINT A.....③**

として、

**PRINT A.....④**

と入力してみよう。今度は「 0」になっちゃったね。


**A=1:RUN.....⑤**

としても「 0」のままだね。

最初に変数Aに1を代入して、次に「PRINT」した時には、「A」の値は「 1」だったのに、プログラムを打ち込んだあと(④の時)には「A」の値は「 0」になっていた。⑤では、「Aは1だよ。」と命令したあとなのに、「A」は「 0」になってしまった。

コンピューターで「A=1」とするためには、「A」という変数の値がいくつか、ということと、その値をどこに記憶しているか、という事の2つを「RAM」と呼ばれる部分に記録しているんだ。この変数の登録を、いつまでも残しておく、次々にいろんな変数を使っていると記憶するための場所がなくなってしまう。そこで、「RUN」命令や、プログラムを新しく追加したり、けずったりした時、それまで保存していた変数の値を、全部一度に消す必要があるんだ。

## クリア機能



この働きのことを「CLEAR (クリア)」機能という。「ホーム・クリアー・キー」が、画面の絵や文字を消すのに対して、この「CLEAR」機能は変数の記憶を消してしまう働きをするんだ。

この「クリアー」機能は、前に出てきた「RUN」やプログラム変更の他にも、次のような命令を使うと自動的に働いてしまうんだよ。

**NEW** (プログラム自体も消してしまう。)

**CLEAR** (プログラムの中で使い、変数の値をクリアする。)

**LOAD** (他のプログラムを読み込む。)

「RUN」命令は、プログラムの中で使うこともあるけど、ほとんどはダイレクトで使う命令だ。それじゃ、「NEW」はどうだろう。


**10 NEW**



と入<sup>にゆうりよく</sup>力したあと、「RUN」させて、リストをとってみよう。今、せ<sup>いま</sup>っかく打<sup>う</sup>ち込<sup>こ</sup>んだ10行は消えてしまうんだ。いたずら<sup>ぎよう</sup>用<sup>き</sup>プログラム<sup>よう</sup>でもない限り<sup>かぎ</sup>、「NEW」命令<sup>めいれい</sup>は、あまりプログラム中<sup>ちゆう</sup>で使<sup>つか</sup>わない方<sup>ほう</sup>がいいみたいだね。

このように、ダイレクトモードで使<sup>つか</sup>う方<sup>こと</sup>が多い命令<sup>おお</sup>には、どんなものがあるんだろう？

## LIST



「LIST」も、その代<sup>だい</sup>表<sup>ひよう</sup>的<sup>てき</sup>な命令<sup>めいれい</sup>だよ。プログラム中<sup>ちゆう</sup>で使<sup>つか</sup>えないこ<sup>こと</sup>ともないが、ほとんどは使<sup>つか</sup>われないね。

LIST (プログラム全部)  
LIST 20- (20行から最後まで)  
LIST -100 (最初から100行まで)  
LIST 20-100 (20行から100行の間だけ)

の4通り<sup>とお</sup>の使<sup>つか</sup>い方<sup>かた</sup>があることは知<sup>し</sup>ってるだろう。


次<sup>つぎ</sup>のプログラムを入<sup>にゆうりよく</sup>力してみよう。

```
10 A=1
20 A=A+1
30 PRINT A
40 GOTO 20
```

このプログラムを「RUN」すると、画面<sup>がめん</sup>には数<sup>すう</sup>字<sup>じ</sup>が一<sup>いち</sup>ず<sup>ず</sup>つふえなが

```
OK
100
200
300
400
RUN
1
2
3
4
5
6
7
8
9
10
11
12
13
14
15
16
17
18
19
20
A=1
A=A+1
PRINT A
GOTO 20
```





ら、次々と表示される。どこかでストップ・キーを押すと、


**BREAK IN ~**

と表示される。「~」はブレーク・キーを押した時に、コンピューターが実行していた行番号だ。ここで、

**PRINT A**

としてみよう。「BREAK ~」と書かれた上の行に表示されている数字と同じ値になっているだろう（タイミングによっては一つ多い値）。

さて、今、中止されているプログラムの実行をその場所から再開したい時には、どうすればいいのだろうか？「RUN」するとまた「1」から数え始めてしまう。




**CONT**

**CONT**

とすると、さっきの数字から表示されるだろう。つまり、ブレーク・キーや、プログラム中の「STOP」という命令で、一度中断されたプログラムを再開するための命令なんだ。

この「CONT」、プログラムを中断したあと、ダイレクト・モードで何かのエラーを出したり、あるいは、プログラムを変更したり、いろんな条件によっては、



**?CC ERROR**

**?CC ERROR**





「キャント・コンティニュー」つまり、「再開できません。」というエラーが出てしまうんだ。

テープへの書き込みや、読み出しの命令、「LOAD」や「SAVE」、また、書き込みがうまくいったかを確かめる「LOAD ?」という命令も、普通はプログラム中では、ほとんど使われないんだ。また、ゲーム・ベーシック画面からBGグラフィック画面を呼び出すための「SYSTEM」命令も、プログラムには必要ないだろうね。

Ver.3では、ダイレクト・モードで使う命令がいっぱい出てきて、ふえてくるよ。ジャンプ編で説明することにしよう。




# 004 -2

## たった 8個の命令で



次のページのリストを見てみよう。「ファミリー・ベーシック」ホップ編で作ったゲーム・プログラムだ。このプログラム、一体いくつの命令を使ってるんだろう？何となくいっぱい使っているようにだけど、実は、全部で8個の命令と4個の関数、それに6個の記号（「AND」とか「=」とか、演算子なんていうむずかしい名前だったね。）しか使っていないんだ。




### 初期設定

- ①画面に表示するキャラクターを決める。
- ②それぞれの最初の位置や、得点の最初の値を決める。
- ③コントローラーからの入力を見て位置を決める。
- ④それぞれのキャラクターを動かす。
- ⑤当たったかどうかの判定。

基本的には、上の5つのことで作られているんだ。

まず、①と②。むずかしい言葉を使うと、初期設定とか、初期値を決める、とかいう。ゲームの一番初めにやることで、ゲームが始まっちゃうと、もうここはいらないんだ。ここで使われているのが、「CHR\$」と「DEF SPRITE」。この2つで表示するキャラクターが決まっているんだ。

### 関数



次に、「STRIG」と「STICK」という2つの関数で、コントローラーのどのボタンを押しているかを調べるのが、③の役割だよ。

コンピューターで言う「関数」とは、たとえば、

**PRINT CHR\$ (122)**

**A=STRIG (0)**

の、「CHR\$」や「STRIG」のように、ある単語をコンピューターに送ると、その中身を数字や文字で返してくる、その単語のことを言うんだ。もし、君のコンピューターに、「サイフ」という命令を送っ

```
10 CLS
20 SPRITE ON
30 DEF SPRITE 0,(0,1,0,0,0)=CHR$(172)+CHR$(173)+CHR$(174)+CHR$(175)
40 DEF SPRITE 1,(0,0,0,0,0)=CHR$(212)
50 'DEF SPRITE 2,(0,1,0,0,1)=CHR$(162)+CHR$(163)+CHR$(160)+CHR$(161)
60 DEF SPRITE 3,(0,0,0,0,0)=CHR$(212)
70 DEF SPRITE 4,(0,1,0,0,0)=CHR$(180)+CHR$(181)+CHR$(182)+CHR$(183)
90 SC=0:SH=3
100 AX=120:BX=120:BY=5
200 IF STICK(0)=1 THEN AX=AX+1
210 IF STICK(0)=2 THEN AX=AX-1
220 IF AX<0 THEN AX=0
230 IF AX>240 THEN AX=240
240 BY=BY+RND(2):BX=BX+RND(9)-4
250 IF BX>240 THEN BX=240
260 IF BX<0 THEN BX=0
270 IF BY>220 THEN BY=5
300 IF FA=1 THEN CY=CY-4
310 IF FB=1 THEN DY=DY+4
320 IF FA=1 AND CY<5 THEN FA=0:SPRITE 1
330 IF FB=1 AND DY>220 THEN FB=0:SPRITE 3
350 IF FA=0 AND STRIG(0)=8 THEN FA=1:CX=AX+4:CY=212
360 IF FB=0 AND RND(6)=1 THEN FB=1:DX=BX+4:DY=BY+16
400 IF FA=1 AND CX+3>BX AND CX+3<BX+15 AND CY+2>BY AND CY+2<BY+15 THEN 1000
410 IF FB=1 AND DX+3>AX AND DX+3<AX+15 AND DY+5>220 AND DY+5<235 THEN 1500
500 SPRITE 0,AX,220
510 IF FA=1 THEN SPRITE 1,CX,CY
520 SPRITE 2,BX,BY
530 IF FB=1 THEN SPRITE 3,DX,DY
700 GOTO 200
1000 FA=0:SPRITE 1:SPRITE 2
1010 SPRITE 4,BX,BY
1020 SC=SC+10
1100 GOTO 2000
1500 FB=0:SPRITE 3:SPRITE 0
1510 SPRITE 4,AX,220
1520 SH=SH-1
2000 LOCATE 10,5:PRINT "SCORE =":SC
2010 IF SH=0 THEN 3000
2020 LOCATE 10,8:PRINT "ノコリ =":SH
2100 BEEP
2110 FOR I=0 TO 1500:NEXT
2120 BEEP:SPRITE 4
2200 CLS:GOTO 100
3000 LOCATE 10,8:PRINT "GAME OVER !"
3100 END
```




たら、サイフの中身がいくらあるかを教えてくれる機能があるでしょう。

#### A=サイフ:PRINT A

とすれば、「300」とか数字で返ってくるはずだね。「0」だって？  
そんなこと、私は知らないよ！）。この「サイフ」というのも立派な関数といえるんだ。

単語のあとに「( )」がついてるのは、たとえば、コントローラーが2台あって、どっちのコントローラーの情報を知りたいのかによって、コンピューターの処理が違うので、コントローラーIなら「(0)」、IIなら「(1)」のように数字を入れて区別するんだ。だから、区別する必要のないものは、単語のうしろの「( )」はいらないことになるね。




#### FRE

#### A=FRE:PRINT A

この「FRE」という関数は、コンピューターのプログラムや変数を記憶する場所（メモリー）が、あと、どの位残っているかを教えてくれるものなんだ。この場合、残りのメモリーというのは1つしかないから、区別する必要がないので、「( )」はいらないのは、もうわかるだろう？

③ではもう一つ、「STRIG」や「STICK」、「RND」で決めた位置が、画面の外に出てやしないかをチェックしているんだよ。自分のキャラクターが、一番左にいて、コントローラーでもっと左に行




け、という命令を出していたら無視する、というようにね。書いてはいけない場所にむりやり書こうとすると、エラーになってしまう。

そうやって決まった各キャラクターの新しい位置を④で書いてある「SPRITE」という命令で、移動したり、見えなくなるものを消したりしているんだ。

⑤では、レーザーが当たったかどうかを判定して、当たっていなければ「GOTO」で③の処理へ進んでいくわけなんだ。


いかにもむずかしそうな長いプログラムも、こうやって分けて行くと、いくつかのかたまりに分けられる。そうして、一つ一つのか





## ルーチン

## ループ



たまりを見ていくと、そのプログラムが<sup>なに</sup>何をしようとしているのか  
が、わかってくるんだ。

このプログラムのかたまりのことを「ルーチン」とも言う。また、  
③→④→⑤→③→④→⑤のように、くり返して<sup>かえ</sup>処理<sup>しり</sup>することを「ル  
ープ」と言う。プログラムのワッ！！っていうわけだね。

①② 初期設定<sup>しよきせつてい</sup>ルーチン

③ 入力判定<sup>はんてい</sup>ルーチン

④ 移動<sup>いどう</sup>ルーチン

⑤ 判定<sup>はんてい</sup>ルーチン

③、④、⑤をメインルーチンまたは、メインループというんだ。  
この「メイン」とは「主要<sup>しゆよう</sup>」という<sup>い</sup>意味<sup>み</sup>だよ。「ルーチン」とか「ル  
ープ」という<sup>たんご</sup>単語<sup>おぼ</sup>は、ぜひ覚えておこう。

# 004-3

## コンピューターに 入力するのは



人間が何をしたいのかを、コントローラーから読みとる関数は、「STRIG」、「STICK」という二つがあるのは勉強したね。それじや、他にはどんなものがあるんだろう？

Ver.2のカセットで電源を入れた直後に、「アナタノナマエハ？」と聞いてきたよね。ということは、コンピューターに数字や文字を教えることができるはずだね。

1Ø A\$=INKEY\$

2Ø CLS:LOCATE Ø,Ø:PRINT A\$

3Ø GOTO 1Ø

このプログラムを入力してみよう。全体の流れはわかるね。A\$という文字変数に「INKEY\$」というのを入れる。20行では、画面を消して、左上にさっきのA\$を表示する。30行では、10行へいく。ストップ・キーを押さない限り、10行から30行までがいつまでも続く、ループ（無限ループ）になっているね。


「RUN」してみよう。画面はまっくらになったね。20行で「CLS」をしているから当然なんだ。ここで何かキーを押してみよう。文字か数字のキーだよ。

すると、何か字が現れたみたいだね。でもすぐ消えちゃった。しかも、現れた文字は、押したキーと同じ字だったみたいだね。


この「INKEY\$」というのは、コンピューターが「INKEY\$」と書かれた行を実行している時に何かキーが押されていたら、そのキーと同じ文字を返してくる文字関数なんだ。

ここで、20行を変えてみよう。

2Ø CLS:LOCATE Ø,Ø:PRINT ASC  
(A\$)




無限ループ




INKEY\$





このまま RUN して、何かのキーを押すと、いろんな数字が出てくる。「A」というキーを押すと「65」と出るはずだ。キャラクターコード表 B を見てみよう。「A」のコードは「65」だ。つまり、「ASC」という命令で押したキーのコードナンバーを表示するプログラムなんだ。


## ASC



ここで、キーボードのカーソル・キーを押してみよう。「▲」のキーを押すと、「30」と表示されるだろう。ところが、キャラクターコード表 B には「▲」なんてのってなくて、B コードナンバー30は、キャラクターテーブル B の D 列の6番目だっけ書いてある。

これは、普通のパソコンと違って、ファミコンではいろんなキャラクターをコンピューター内部に持っているため、コードナンバー0から31の間は2通りの使い方をしているせいなんだ。誰かがパソコンをもっていたら、そのマニュアルのキャラクターコード表を見てもらおう。コードナンバー30（10進法の）は「↑」となっているパソコンがほとんどだ。

## 特殊キーのコード



この「INKEY\$」は、あとで説明する命令と違って「▲」などのカーソル・キーや「INS」、「DEL」キーなどの特殊キーが押されたかどうかをわかる命令なんだ。この命令を使うと、マリオなどのキャラクターをコントローラー抜きで、キーボードのカーソル・キーで動かすことができるんだよ。

```

10 AX=0:BX=0
20 IF STICK(0)=1 THEN
  AX=AX+1
30 IF INKEY$=CHR$(28)
  THEN BX=BX+1
40 PRINT AX,BX
50 GOTO 20

```

このプログラムは、コントローラーの「+」を右へ押すとAXが、キーボードの「▶」を押すとBXが、1ずつふえていく、というものなんだ。キーボードがコントローラーの代わりをできるのがよくわかるだろう。

「◀」⇒CHR\$(29)

「▲」⇒CHR\$(30)

「▼」⇒CHR\$(31)

いろんな特殊キーをためしてみよう。

㊦ ページのプログラムを実行している時にファンクションキーを押すと、そのファンクションキーに登録されている文字が1字ずつ表示される。これは、ファンクションキーを押すことで、一度に文字が並んで行列まちをしていると思ってほしい。

パソコンによっては、この「INKEY\$」が現われるまでに押したキーを記憶していて、この行列まちのような形で覚えているもの（キー・バッファという）もある。

では、2字以上の文字や数字を入力するには、どうすればいいんだろう？ 前のプログラムを「NEW」して消してから、次のプログラムを入力しよう。


```

10 INPUT A$
20 PRINT A$

```


「RUN」すると「？」マークが出て、すぐうしろにカーソルが点

004 少しうしろをふり返ってみよう 21



INPUT





めつ  
減しているね。


「F」, 「M」, 「I」と一つづつキーを押してみよう。一字づつ画面  
に表示され、「I」のうしろにカーソルが移っただろう。ここでリター  
ン・キーを押すと、次の行に、もう一度「FMI」と表示されたね。

2回目の「FMI」はわかっただろう。20行の「PRINT A\$」と  
いう命令だね。ということは、A\$の中身は「FMI」になっている。  
つまり、10行の「INPUT A\$」は、「A\$」の中身は何にしましょ  
う？とコンピューターから君に問いかけるようにしろ、という命令  
だったんだ。


「INPUT」という命令を受けると、コンピューターはまず「？」  
というマークを出し、うしろにカーソルを点滅させることで、早く  
入力して、と要求してくるんだ。

もう一度「RUN」して、今度は「▼」と「K」を一回づつ押して  
みよう。「▼」キーを押すとカーソルは一段下がって、下に「K」を  
表示した。でも、リターンを押すと、A\$には「K」の一字しか入  
っていないことがわかる。つまり、「INKEY\$」と違って特殊キーは  
受けつけないのがわかるよね。

「RUN」したあと、何でもいから文字キーを32個以上押したあ  
と、リターンキーを押すと、



## ? ST ERROR



## ? ST ERROR IN 10

となる。ホップ<sup>へん</sup>編<sup>はな</sup>で話<sup>も</sup>した文字列<sup>じれつ</sup>の長さ<sup>なが</sup>制限<sup>せいげん</sup>を覚えてるかな? このエラーは「ストリング・トゥー・ロング」エラー<sup>おぼ</sup>といって、文字列<sup>も</sup>の長さ<sup>も</sup>が長すぎるよ、というエラーなんだ。

10 INPUT A

20 PRINT A

と変えてみよう。前の命令<sup>まえ めいれい</sup>では「A\$」<sup>こんど</sup>、今度は「A」<sup>すうち</sup>。これは数値<sup>すうち</sup>を入<sup>い</sup>力<sup>りき</sup>しろという命令<sup>めいれい</sup>なんだ。ためしに「1」のキーを押すと「1」と表示<sup>ひょうじ</sup>される。前のプログラムの時<sup>とき</sup>では「1」を入<sup>い</sup>力<sup>りき</sup>すると「1」と表示<sup>ひょうじ</sup>されたはずだね。違い<sup>ちが</sup>がわかるかな? 前のプログラムでは「1」は文字<sup>も</sup>としてあつかったため、そのまま「1」<sup>ひょうじ</sup>と表示<sup>ひょうじ</sup>されたけど、今度は数字<sup>こんど</sup>としてあつかうために、「1」の手前<sup>てまえ</sup>に「+」を書<sup>か</sup>いてあるためなんだ。[(「+」の場合は省<sup>は</sup>略<sup>りゃく</sup>して代わりに、「」(スペース)を置く約束<sup>やくそく</sup>だったよね。)]

「F」, 「M」, 「I」を押<sup>お</sup>してリターンキー<sup>お</sup>を押<sup>お</sup>してみよう。数値<sup>すうち</sup>変<sup>へん</sup>数<sup>すう</sup>の中身<sup>なかつみ</sup>を聞<sup>き</sup>く命令<sup>めいれい</sup>に文字列<sup>も</sup>を入<sup>い</sup>力<sup>りき</sup>するんだから、ほとんどのパソコンでは「Type Mismatch」(タイプが違<sup>ちが</sup>うよ!) エラーになるけど、ファミコンではエラーにならないで、何も入<sup>な</sup>力<sup>に</sup>しない(つまり0) あつかいになるようだね。



```
OK
10 INPUT "Xハ";A
20 PRINT A
RUN
Xハ?
```

10 INPUT "Xハ";A

と変えてみよう。

Xハ?

となって「?」のうしろにカーソルが点滅している。

プログラムの中でいくつも「INPUT」文があると、入力を要求されたときに、何を入力すればいいのか、わからなくなることがある。そんなとき、「INPUT」のあとに「」(ダブルクォーテーションマーク)で囲んだ文字列を書いて、「;」(セミコロン)を置くと、まず文字列の中を表示したあと、入力を要求してくるんだ。

10 PRINT "Xは";: INPUT A


というのと同じ働きなんだ。

今度は、


10 INPUT "X=",A

としてみよう。今度は「X=」のうしろで、直接カーソルが点滅しているね。「?」のマークが出ないようにするには、文字列のあとの「;」を「,(カンマ)に変えるだけでいいのだ。ただ、この使い方は、必ず文字列がないと使えないよ。つまり、この10行を書きかえて、

10 PRINT "X=";: INPUT ,A



?と,



```

OK
100 INPUT "X=",A
RUN PRINT A
X=

```

ではだめで、

```

10 PRINT "X="; : INPUT " "
    , A

```

としなければいけない。

なぜだろう？この「,」は、実はいくつもの変数を入力するときの区切り記号にも使うんだ。

```

10 INPUT "A, Bは"; A, B
20 PRINT A*B

```

としてみよう。「RUN」したあと、「3」キーとリターンキーを押してみよう。あれ、もう一度「？」が出てきたね。「4」キーとリターンキーを押すと、「12」と表示される。

```

PRINT A:PRINT B


```

としてみると、「A」には「3」、「B」には「4」が入っている。10行の命令は一度に「A」、「B」2つの変数の値を入力せよ、という意味だったんだ。だから、最初に「3」を入力しても、もう一個いるから、もう一度「？」のマークを出して「B」の方の値の入力を要求してきたわけなんだ。

もう一度「RUN」して、今度は「2」、「,」、「5」、リターンキーを押してみよう。「10」になっただろう。入力する時も「,」で

004 少しうしろをふり返ってみよう 25

区切り記号





```

OK
10 INPUT "A,B":A,B
20 PRINT A*B
RUN
A,B:3
?4
12
OK

```

```

OK
RUN
A,B:2.5
10
OK

```

区切ることによって、2個の数字を一度にできるんだ。

この「,」で区切ることによって2個以上、それも、文字も数字も入力できるんだ。

**INPUT A,B\$,C**

という命令なら「3」、「,」、「T」、「E」、「,」、「4」と入力すると、ちゃんとそれぞれの変数に、押したものが入っている。

```
10 INPUT "A,Bハ ";A,B
```

```
20 PRINT "A+B=";A+B
```

```
30 PRINT "A-B=";A-B
```

```
40 PRINT "A*B=";A*B
```

```
50 PRINT "A/B=";A/B
```

このプログラム、もう説明しなくても、君にはわかるよね？

最後に、じゃあ「ファミ、コン」というように「,」をふくんだ文字列は絶対入力できないじゃないか！と気がついた君。エライ！！その通りだ。そのかわり、


```
10 LINPUT A$
```

```
20 PRINT A$
```

というプログラムを実行してみよう。ちゃんと「,」の入った文字列も一つの文字列として入力できたね。

```
10 LINPUT "TEST=";A$
```

としてみよう。「A」と入力すると、今度は、A\$は「TEST=A」という文字列になっている。この「LINPUT」のあとに文字列をつけると、その文字列と、入力された文字列をたしたものが文字変数に代



**LINPUT**

```
OK
10 INPUT "TEST=" ; A$
20 PRINT A$
RUN
TEST=■
```

にゆう  
入される，という仕組みになっているんだ。

だから，この「INPUT」は，数値変数の入 力や2つ以上の変数  
の入 力には使えないことになるね。つまり，文字変数だけの入 力  
用なんだ。また，この命令を使うと「？」は表示されない。だから  
「"」で囲んだ文字列のうしろは，「；」でも「，」でも働きはま  
ったく一諸なんだ。



# 作戦指令 005

ベーシック  
自由自在



# 005-1

## あっちの道、 こっちの道



ファミコンの「メモリー」が<sup>すく</sup>少ない、ということは、ゲーム<sup>つく</sup>作りには一番の難関になるのはわかるだろう。わずか1982バイトの<sup>なか</sup>中で、よりおもしろいゲーム<sup>つく</sup>を作るためには、いろんなテクニック<sup>つか</sup>を使わなければならないんだ。

```
10 DEF SPRITE 0, (0, 1, 0, 0, 0)
   =CHR$(172)+CHR$(173)+CHR$(174)+CHR$(175)
```

```
90 DEF SPRITE 4, (0, 1, 0, 0, 0)
   =CHR$(180)+CHR$(181)+CHR$(182)+CHR$(183)
```

このプログラムは、ホップ<sup>へん</sup>編のゲームリストの一部<sup>いちぶ</sup>だね。よく見ると、10行と90行で違うのは「DEF SPRITE」の次の数字が「0」と「4」というのと、「CHR\$」のカッコの中<sup>なか</sup>だけだ。

このプログラムを打ち込んだところで、ダイレクトに


**PRINT FRE**

と入<sup>にゅうりよく</sup>力してみよう。「FRE」はメモリーの残り<sup>のこ</sup>を表わす関数<sup>かんすう</sup>だったよね。「1860」になったはずだね。とすると、この2行<sup>ぎよう</sup>のプログラムで「122」バイト（メモリーの単位<sup>たんい</sup>をバイトという。）使ったことになるね。

さて、「NEW」したあと、次のプログラムを打ち込んでみよう。

```
10 A=0:B=172:GOSUB 5000:A=
   4:B=180:GOSUB 5000
```

```
5000 DEF SPRITE A, (0, 1, 0, 0,
   0)=CHR$(B)+CHR$(B+1)+CHR$(B+2)+CHR$(B+3):RET
```



**GOSUB**



## URN

この状態で「FRE」をしらべると、「1878」になるだろう。つまり、こっちのプログラムでは「104」バイトしか使っていないわけだ。使うメモリーが少なくなったのはいいけど、さて、この2つのプログラムは同じ働きをするのかな？

100 SPRITE 0, 100, 100

110 SPRITE 4, 150, 100


120 END

上のプログラムをそれぞれのプログラムにつけ加えて「RUN」すれば、同じ働きをすることがすぐわかるね。

じゃあ、2番目のプログラムはどういう働きをしているんだろう？

10行で変数A, Bにそれぞれ0, 172を代入したあと「GOSUB 5000」と書いてある。今まで「GOTO」という命令が出てきたことはあったけど、これは初めてだね。

コンピューターがプログラムの上を歩いている人としよう。その人が歩いている足の下にある命令を実行しているというわけだ。こ



の<sup>ひと</sup>人が「GOTO」の命令に<sup>めいれい</sup>会<sup>あ</sup>うと、そのあとに書かれた行番号<sup>ぎょうばんごう</sup>の所<sup>ところ</sup>へ飛んでいってしまう。「GOTO 1000」と書かれていたら、今<sup>いま</sup>いる所<sup>ところ</sup>から、行番号1000に書かれている命令<sup>めいれい</sup>の上<sup>うへ</sup>までジャンプしていくわけなんだ。

それに対して「GOSUB」はどうだろう？ 行番号100に「GOSUB 1000」という命令<sup>めいれい</sup>があると、1000行<sup>ぎょう</sup>の命令<sup>めいれい</sup>から実行<sup>じつこう</sup>しはじめる。しかし、途中<sup>とちゆう</sup>で「RETURN」という命令<sup>めいれい</sup>に行きあうと、もとの「GOSUB 1000」という命令<sup>めいれい</sup>の次<sup>つぎ</sup>に書かれた命令<sup>めいれい</sup>の所<sup>ところ</sup>に帰<sup>かえ</sup>ってくるんだ。

言<sup>い</sup>いかえれば、コンピューターは行番号100で「GOSUB」命令<sup>めいれい</sup>に会<sup>あ</sup>うと、本体<sup>ほんたい</sup>はそこに残<sup>のこ</sup>したまま、ダミー<sup>だみり</sup>（代理<sup>だいり</sup>）を1000行<sup>ぎょう</sup>に飛<sup>と</sup>ばす。そして「RETURN」に会<sup>あ</sup>うと、そのダミー<sup>だみり</sup>は本体<sup>ほんたい</sup>の所<sup>ところ</sup>へ帰<sup>かえ</sup>ってきて、また、本体<sup>ほんたい</sup>が歩き始めるわけなんだ。

このように「GOSUB」命令<sup>めいれい</sup>は、出<sup>で</sup>て行<sup>い</sup>った所<sup>ところ</sup>から必<sup>かなら</sup>ず「RETURN」命令<sup>めいれい</sup>で帰<sup>かえ</sup>って来<sup>こ</sup>なければいけ<sup>い</sup>ないんだ。

10 I = 0

20 GOSUB 1000


30 END

100 I = I + 1 : GOTO 20


このプログラムを「RUN」すると、すぐ「? OM ERROR」になる。メモリー不足<sup>ふそく</sup>エラー、という意味<sup>いみ</sup>だね。ところ<sup>ところ</sup>が、「PRINT FRE」とすると、「1938」バイトも残<sup>のこ</sup>っているよ、と表示<sup>ひょうじ</sup>されるね。

プログラムを<sup>み</sup>見ると、20行<sup>ぎょう</sup>から「GOSUB」命令<sup>めいれい</sup>で100行<sup>ぎょう</sup>へいったのに100行<sup>ぎょう</sup>からは「RETURN」命令<sup>めいれい</sup>でなく「GOTO」命令<sup>めいれい</sup>で20行<sup>ぎょう</sup>へいっている。そして、その20行<sup>ぎょう</sup>から、また「GOSUB」で100行<sup>ぎょう</sup>へいくことをくり返<sup>かえ</sup>しているね。

コンピューターが「GOSUB」命令<sup>めいれい</sup>でどこかへ行<sup>い</sup>くときには、返<sup>かえ</sup>てくるためにも、どこから出<sup>しゅつぽつ</sup>発<sup>はつ</sup>したか（GOSUB 命令<sup>めいれい</sup>がどこに書かれていたか）を記憶<sup>きおく</sup>する必要がある<sup>ひつよう</sup>んだ。ところがこのプログラムでは何回<sup>なんかい</sup>も「GOSUB」をくり返<sup>かえ</sup>しているため、出<sup>しゅつぽつ</sup>発<sup>はつ</sup>点を覚<sup>おぼ</sup>える場所<sup>ばしょ</sup>がなくなっちゃったんだ。ためしに「PRINT I」をしてみよう。29回<sup>かい</sup>くり返<sup>かえ</sup>したことがわかるだろう。




RETURN



どこから出発したか



## CLEAR PRINT FRE




サブルーチン

としてみると、1942バイトだね。(変数 I をクリアーしたから 4 バイトふえた。)とすると、この出発点を覚えているメモリーは、僕達の使える部分とは別に、コンピューターが持っているのがわかるね。さっきの「OM」エラーとは、その部分がメモリー不足になったよ、という意味だったんだ。

このように、「GOSUB」で行って「RETURN」で帰ってくる間(ダミーが実行している部分)を「サブ・ルーチン」と呼ぶんだ。前のプログラムの 5000 行がそれにあたるわけだね。このサブルーチンの中から別のサブルーチンへ行ったりすることもできるんだよ。図 1

●図 1



を見てみよう。「サブルーチンへ行くことを、ふつう、「GOTO 100」が「100 行へいく」に対して、「GOSUB 100」は「100 行のサブルーチンを呼ぶ」または、「サブルーチンコール」と言う。

これで 2 番目のプログラムの役割がわかったよね! 120 行の「END」はどうだろう。もしこの「END」がなかったら、コンピューターの動きは次のようになってしまう。

10 行 → 5000 行 → 10 行 → 5000 行 → 100 行 → 110 行 → 5000 行

そして、ここで「? RG ERROR」(=リターン・ウィズアウト・ゴーサブ)、つまり、「GOSUB」命令を受けてないのに「RETURN」があるよ、というエラーが出てしまう。サブルーチンをプログラムのうしろの方に置くとときによくやるエラーなんだ。プログラムの終わりには、必ず「END」をつけるくせをつけようね。

このように、少し気をつけなきゃならない点はあるけども、サブルーチンは使い方がさううまくすれば、プログラムを小さくするのに

やくだ  
役立つし、わかりやすくプログラムを組むことだってできるんだよ。  
こんかい  
今回の例では、わずか18バイトしか小さくならなかったけど、同じ  
しより  
処理をくり返すほど、サブルーチン形式にした方が使うメモリーは  
すく  
より少なくてすむんだ。


③ 必しも、「GOSUB」1個に「RETURN」1個が必要というわけ  
ではない。

たとえば、図2 では、「GOSUB」3個に「RETURN」2個だし、  
たとえば、

```
10 A=0:GOSUB 100
20 END
100 IF A=1 THEN RETURN
110 A=A+1:RETURN
```

というプログラムでも間違いではないんだ。命令の流れで  
「GOSUB」で行なった所からは必ず「RETURN」に行き会えれば  
それでよいのだ。


●図2





# 005-2

## くり返して 何回も




### §1 思い出したくないテストの点数

ホップ編のプログラムのうち、

```
211Ø FOR I=Ø TO 15ØØ:NEXT
```

というのを覚えているかな？ この行はただの時間待ちの行だと説明したよね。

PAUSE



```
211Ø PAUSE 7Ø
```

これでも大体同じくらいの時間待ちをしてくれるんだ。

「PAUSE」という命令には2つの使い方があるんだよ。

PAUSE

とだけすると、何かのキー入力があるまでは、ずっと待っているし、

```
PAUSE 1ØØØ
```

とすると、約11秒間待って次の命令に進む。

```
1Ø BEEP
```

```
2Ø PAUSE 1ØØØ
```

```
3Ø BEEP
```

```
4Ø PAUSE
```


```
5Ø BEEP
```

「PAUSE」のあとに続く数字は0~32767の間が許されている。それ以外の数字だと「? IL ERROR」になってしまうんだ。

じゃあ、もとのプログラムの2110行では約1秒弱の間、何をしていたんだろう？


実は、変数Iに0から1つずつ1500まで1501回代入していたんだ。おふろに入ってる時、100まで数えて、っていうあれと同じことをしてたんだなー。

PAUSE 200



“時間がたつと青になる”

PAUSE



“押しボタンを押さないと赤のまま”

それじゃあ、この命令は何の役にも立たないじゃないか。ところがどっこい、すごく役に立つ命令なんだ。

ここに、君の国語と社会と理科の過去3回分のテストがあるという。

	こくご 国語	しやかい 社会	りか 理科
1回目	60	70	50
2回目	30	40	80
3回目	90	100	20

各回の平均点と科目別の平均点を出すには、どんなプログラムを作ればいいんだろう？




リスト1

```
100 K1=60:K2=30:K3=90
110 S1=70:S2=40:S3=100
120 R1=50:R2=80:R3=20
200 PRINT "コクゴ"="(K1+K2+K3)/3
210 PRINT "シャカイ"=(S1+S2+S3)/3
220 PRINT "リカ"=(R1+R2+R3)/3
300 PRINT "1カイメ"=(K1+S1+R1)/3
310 PRINT "2カイメ"=(K2+S2+R2)/3
320 PRINT "3カイメ"=(K3+S3+R3)/3
```

K, S, R はそれぞれ<sup>こくご</sup>国語, <sup>しゃかい</sup>社会, <sup>りか</sup>理科の<sup>かしらもじ</sup>頭文字で, そのあとの, 1, 2, 3 は<sup>かいめ</sup>1回目, <sup>かいめ</sup>2回目, <sup>りやく</sup>3回目の略だ。このプログラムはたしかに間違っていない。

でも, もしこれが<sup>かいぶん</sup>10回分のテストの<sup>けうか</sup>結果という<sup>こと</sup>事になるとどうなるだろう?

K1とK10



100 K1=60: ~ : K10=80

ファミコンでは変数の名前は最初の2文字で判断していたから, 「K1」と「K10」は<sup>おな</sup>同じ変数と<sup>み</sup>見られてしまうね。じゃあ, <sup>なまえ</sup>名前をかえて,

100 KA=60: ~ : KJ=80

<sup>ぜんぶ</sup>全部のプログラムをこの<sup>ちようし</sup>調子で書くのは, <sup>たいへん</sup>ちょっと大変だし, プログラムが<sup>おお</sup>大きすぎちゃって, メモリーがたりなくなっちゃう。そこで<sup>あた</sup>新しい<sup>かんが</sup>考え方を<sup>かた</sup>してみよう。

リスト2

```
100 DIM K(3),R(3),S(3),T(3)
110 FOR I=1 TO 3
120 READ K(I),R(I),S(I)
130 NEXT
200 DATA 60,70,50,30,40,80,90,100,20
300 FOR I=1 TO 3
310 KH=KH+K(I)
320 SH=SH+S(I)
```

```

330 RH=RH+R(I)
340 T(I)=K(I)+S(I)+R(I)
350 NEXT
500 PRINT "コフコ"=;"KH/3
510 PRINT "シャカイ"=;"SH/3
520 PRINT "リカ"=;"RH/3
600 PRINT "1カime"=;"T(1)/3
610 PRINT "2カime"=;"T(2)/3
620 PRINT "3カime"=;"T(3)/3

```

## §2 READとDATA

あたらしい単語がいくつも出てきたね。「DIM」とか「READ」とか「DATA」、それに「K (3)」とか「R (3)」とかは、何だろう？

```

10 A=0
20 READ A
30 PRINT A
40 DATA 5

```

このプログラムを「RUN」してみよう。「5」と表示されたね。

「DATA」というのは、「資料」という意味の英語「DATUM」の複数形なんだ。データベースとかよく聞くと思うけどその「データ」なんだよ。

「READ」は「読む」という意味だね。とすると、20行は、データを読みについて、そのデータを「A」に入れる役割をしているんだ。

```
READ (DATA): A = (DATA)
```


とも書ける役割をしている。

```


10 READ A,B,C$
20 PRINT A:PRINT B:PRINT C$
30 DATA 3,5,"TEST"

```

ちゃんと「A」には「3」、「B」には「5」が入ってるのがわかる



DATA



READ



```

OK
10 READ A,B,C$
20 PRINT A:PRINT B:PRINT C$
30 DATA 3,5,"TEST"
40 RUN
TEST
OK

```

だろう。しかも「C\$」には、ちゃんと「TEST」の文字が<sup>もじ</sup>入<sup>はい</sup>っている。「READ」命令<sup>めいれい</sup>があるとコンピューターは「DATA」と書<sup>か</sup>かれて<sup>きよう</sup>いる行<sup>ぎよう</sup>を探<sup>さが</sup>しに<sup>つぎ</sup>いって読み込<sup>よこ</sup>む。次の「READ」のときは、さっき読<sup>よ</sup>んだ「DATA」の次<sup>つぎ</sup>のものを読み込<sup>よこ</sup>みに行く。したがって「DATA」が書<sup>か</sup>かれた行<sup>ぎよう</sup>は、「READ」文<sup>ぶん</sup>の前<sup>まえ</sup>にあってもいいし、また、「READ」文<sup>ぶん</sup>と「DATA」文<sup>ぶん</sup>が何個所<sup>なんかしよ</sup>にわかれていても、基本的に「READ」と「DATA」の数<sup>かず</sup>さえ合<sup>あ</sup>って<sup>とき</sup>いけば、それでいいわけなんだ。

じゃあ、「DATA」の数<sup>かず</sup>が合<sup>あ</sup>ってない時はどうなるのかな？

10 READ A, B


リスト 3

```

100 DATA 10
200 READ A,B
300 READ C
400 DATA 3,2,1
500 READ D,E
600 DATA 6


```

# READ A READ B READ C



## 20 DATA 1

### ?OD ERROR



この場合は「?OD ERROR」(アウト・オブ・データ)つまり、データの数が足りないよ、というエラーが出てしまうんだ。

データの方の数が多い場合には、エラーにはならないけど、ふつうは、必要以上にデータをわざわざ書くことはないだろうね。

## 10 READ A\$, B

## 20 DATA 1, 1

この場合、文字変数「A\$」は「1」を文字としてとり入れ、数値変数「B」は数値として「1」をとり入れる。ということは、文字変数用の「DATA」は「"」マークで囲まなくてもいいことになるね。


## 20 DATA TEST, 1

でもいいわけだ。じゃ、

## 20 DATA TEST, TEST2

としてみよう。

### ?SN ERROR



## ?SN ERROR IN 10


となったはずだね。これは、数値変数「B」を読みに行ったのに、そこには数値でなく「TEST2」という文字列しかない、というエラーなんだ。ほんとうは「DATA」文の書きまちがいかも知れないのに、エラーとしては必ず「READ」文のある行のものとしてあつかわれるので、要注意だぞ！

このように、「READ」文の中の文字変数と数値変数との順番と「DATA」文の中の文字列と数値との順番とは、いつも同じでな



## 文字変数

## 数値変数



ければいけないんだ。


もう一つ、ついでに新しい命令<sup>あたらしいめいれい</sup>を覚えちやおう。

- 1Ø READ A, B, C
- 2Ø RESTORE 7Ø
- 3Ø READ D, E, F\$
- 4Ø RESTORE 7Ø
- 5Ø READ G, H, I\$, J
- 6Ø DATA 4
- 7Ø DATA 1, 2, TEST, 3


このプログラムで、「B」、「D」、「G」は1、「C」、「E」、「H」には2、「F\$」、「I\$」にはTEST、「J」に3か、「A」に4か<sup>はい</sup>入る。

「READ」文<sup>ぶん</sup>が1回あるごとに、「DATA」文<sup>ぶん</sup>の何番目を読むか、ということをコンピューターは覚えてるんだけど、この「RESTORE XX」という命令<sup>めいれい</sup>があると、XX行<sup>ぎょう</sup>からあとにある、「DATA」文<sup>ぶん</sup>から読み込み<sup>よみこみ</sup>はじめる。というわけなんだ。当然、この「XX」が存在<sup>ざい</sup>しない行<sup>ぎょう</sup>だと「? UL ERROR」(アンディファインド・ライン・ナンバー) そんな行<sup>ぎょう</sup>、みつからないよ、というエラーになるし、「XX」行<sup>ぎょう</sup>からうしろに「DATA」文<sup>ぶん</sup>がなければ「? OD ERROR」になってしまうんだ。


## RESTORE




㊦ 作戦資料001 バグ情報<sup>じやうほう</sup>を見て!



RESTORE 70



RESTORE 70





### § 3 配列変数とDIM

リスト2にもどろう。リスト1とくらべてみると、どうも「K(3)」  
というのは、K1, K2, K3のことらしい。

```
10 FOR I=1 TO 3
20 A=A+1
30 NEXT
```


.....(A)

この命令は、

```
10 A=A+1
20 A=A+2
30 A=A+3
```

.....(B)

というのと同じ意味をもつ。「1」づつふえていく数をたしてる  
んだ。ところが、1から3までだと同じ3行ですむけど、1から100  
までだと⑥の形では100行にもなる。けれども、④の形式では同じ3  
行ですむ。



1Ø I=1

2Ø A=A+1

.....©

3Ø IF I<3 THEN I=I+1

:GOTO 2Ø

としても、①と同じプログラムにある。つまり、「FOR ~ TO ~ NEXT」という文型は、ある数字、この場合は「I」（これを制御変数という。）を、ある値から、ある値まで変化させ（この場合1から3）で、いろんな処理を行なう命令なんだ。

1Ø FOR I=1 TO 99 STEP 2

2Ø A=A+1

3Ø NEXT

これは、「1」から「99」までの奇数だけの数を、たす、というプログラムだ。「STEP 2」というのは「I」を2つずつふやしていく、という意味だ。つまり、

FOR I=1 TO 3

というのは、

FOR I=1 TO 3 STEP 1

の省略形だったんだ。「STEP」をつけなければ「STEP 1」のことと一緒にのことなんだよ。当然、

FOR I=8Ø TO 2 STEP -2

のようなのもできるんだ。

FOR I=1Ø TO 5 STEP 2

はどうだろう。まず「I=1Ø」として処理を行なう。「NEXT」と出会うと「I=12」になる。ところが「TO 5」とあって、「I>5」だから、くり返しはしないで次の命令にうつる。つまり、エラーにはならない。（©にあてはめてごらん。）

リスト2がわかってきたかな？「FOR NEXT」で「I」は1から3まで変化する、とすると「K(I)」というのは、K(1), K(2), K(3)と3回出てくるわけだ。リスト1のK1, K2, K3と一緒に緒だね。

図3を見てみよう。この「K( )」のように書くと、メモリーが許すかぎり、いくつふえても同じ「K」という名前の変数の仲間として使えるんだ。



リスト1のやり方では、回数<sup>かいすう</sup>がふえてくると、どうしようもないのは前に言った通り<sup>まえにい</sup>なんだけども、この「( )」を使った変数<sup>へんすう</sup>のやり方<sup>かた</sup>だと、たとえば、回数<sup>かいすう</sup>が10回なら、

```
110 FOR I=1
TO 10
```

と直して、あとは200行<sup>ぎょう</sup>のDATA文<sup>ぶん</sup>をそれに応じてふやすだけでいいんだ。

このような「( )」を使った変数<sup>へんすう</sup>を「配列変数<sup>はいれつへんすう</sup>」というんだ。ただ、この配列変数<sup>はいれつへんすう</sup>を使うときは、あらかじめ、「こんな配列変数<sup>はいれつへんすう</sup>を使いますよ。」とプログラムのはじめでコンピューター<sup>たい</sup>に対して、「宣言<sup>せんげん</sup>」しておかなければいけないんだ。

「DIMENSION」<sup>ようせき</sup>、容積<sup>ようせき</sup>とか大きさ、次元<sup>じげん</sup>という意味の略語<sup>りやくご</sup>の「DIM」というのが、その宣言文<sup>せんげんぶん</sup>だったんだ。

```
DIM K(3)
```

これは、「K」という文字<sup>もじ</sup>であつかう配列変数<sup>はいれつへんすう</sup>を、「K(0)」から「K(3)」まで使いますよ、という意味<sup>いみ</sup>なんだよ。(カッコの中の数字<sup>なかに</sup>を添字<sup>すうじ</sup>と添字<sup>てんじ</sup>という。)

この「DIM」で宣言<sup>せんげん</sup>してない配列変数<sup>はいれつへんすう</sup>を使ったり、宣言<sup>せんげん</sup>している大きさより大きい添字<sup>すうじ</sup>を使ったりすると、

```
? IL ERROR
```

になってしまう。つまり回数<sup>かいすう</sup>が10回になった時は、100行<sup>ぎょう</sup>も、

```
100 DIM K(10), R(10), S(10), T
(10)
```

●図3

K (1)

K (0)

K (1)


K (2)

K (3)


K (4)

⋮

## 配列変数



## DIM



なお ひつよう  
に直す必要があるわけだ。

リスト2の「T(3)」は各回の合計、KH, SH, RHはそれぞれ国  
・社・理の合計だってことはわかるよね。リスト2では2回も  
「FOR NEXT」ループを使っている。これを短かくすると、

●  
リス  
スト  
4


```
100 DIM K(3),R(3),S(3),T(3)
110 FOR I=1 TO 3
120 READ K(I),R(I),S(I)
130 KH=KH+K(I)
140 SH=SH+S(I)
150 RH=RH+R(I)
160 T(I)=K(I)+S(I)+R(I)
170 PRINT I;"カイメ=";T(I)/3
180 NEXT
200 DATA 60,70,50,30,40,80,90,100,20
500 PRINT "コクゴ"=;KH/3
510 PRINT "シャカイ"=;SH/3
520 PRINT "リカ"=;RH/3
```

になる。いずれにしても、リスト1よりはメモリーをたくさん使う  
けど、あつかうデータの量が多くなると、今度はリスト2や4の方  
が、断然小さくなってくるのはわかるよね。

ところで、ページでも言ったように、テストの点数はたてと横に  
書ければわかりやすい。配列変数も図3 だけじゃなく、たてと横  
に使えるんだらうか？

図4 をみてみよう。

2次元



ファミコンでは、配列変数は2次元まで許されている。1次元とい  
うのは、図3 のような、たて、または横だけの配列変数なんだ。  
このように、たてと横の2つの要素をもったものを2次元というん  
だ。カーソルの位置を表わすのも、X座標・Y座標という2つの要  
素が必要だったよね。あれと同じっていうわけなんだ。

さっそく、それを使ってプログラムを書いてみよう！



●図 4

	回数 別 合 計	国 語	社 会	理 科	
科目別合計	A (0.0)	A (1.0)	A (2.0)	A (3.0)	
1 回目	A (0.1)	A (1.1)	A (2.1)	A (3.1)	
2 回目	A (0.2)	A (1.2)	A (2.2)	A (3.2)	
3 回目	A (0.3)	A (1.3)	A (2.3)	A (3.3)	

●  
リス  
スト  
5

```

100 DIM A(3,3)
110 FOR I=1 TO 3:FOR J=1 TO 3
120 READ A(I,J)
130 A(0,I)=A(0,I)+A(I,J)
140 A(J,0)=A(J,0)+A(I,J)
150 NEXT
160 PRINT I;"カイメ=";A(0,I)/3
170 NEXT
200 DATA 60,70,50,30,40,80,90,100,20
300 FOR I=1 TO 3
310 READ A$
320 PRINT A$;"=";A(I,0)/3
330 NEXT
400 DATA コフコ",シャカイ,リカ

```

コンピューターは、いつも、「0」からはじまることは何回も言っ  
 たよね。「DIM A (3, 3)」というのは、A という配列変数は、た  
 て4×横4、の16通りを使える、というわけだ。

図4 とリスト5を見くらべながら、このプログラムがどうい  
 うことをしているかは、もう、わかるよね。






# 005-3

## 文字と数字の あいだから


「ASC」や「CHR\$」はもう知ってるね。文字と数字は、いろいろな形でつながっている。また、コンピューターの中にはいろんな関数がある。それらをサンプルを使いながら紹介しよう。

リスト6

```
100 A$="TESTPROGRAM"
110 B$=LEFT$(A$,4):PRINT B$
120 B$=RIGHT$(A$,7):PRINT B$
130 B$=MID$(A$,5,3):PRINT B$
200 C=LEN(A$):PRINT C
```



LEN



LEFT\$

このプログラムに新しく出てきた単語は、文字列をいろいろ動かす命令なんだ。

一番最後の「LEN」というのは、文字列の長さを返してくれる。うしろの「( )」の中は、プログラムのように文字変数でもいいし、

**C=LEN("TESTPROGRAM")**

のように、直接、文字列を入れてもいい。これは、他の3つの命令も同じだ。

**B\$=LEFT\$("TESTPROGRAM",4)**

とも書けるわけだ。

110行はカッコ「( )」の中の文字列のうち、左側から4つの文字をB\$とする、という命令なんだ。もちろん「、」のあとの「4」を「5」にすれば、5つの文字になることはわかるね。

120行は逆に、右側の7文字をB\$にする命令だ。

A\$=TESTPROGRAM

B\$=LEFT\$(A\$,4)

C\$=MID\$(A\$,6,2)

D\$=RIGHT\$(A\$,3)



A\$=TEST PROGRAM


B\$=TEST

C\$=RO

D\$=RAM

130行は、文字列A\$の左側から5番目の文字から、7番目の文字までの3文字を、B\$に入れるという意味なんだ。

## MID\$



MID\$ (文字列, 先頭の文字は左側から数えて何番目か, いくつの文字数か)

というふうにかく。

とすると、110行、120行はそれぞれ、

110 B\$=MID\$(A\$,1,4)

120 B\$=MID\$(A\$,5,7)

と書いても同じになるよね。これらはそれぞれ英語の、

RIGHT	=	右
LEFT	=	左
MIDDLE	=	中, 中央
LENGTH	=	長さ

の略字なんだ。




## ●リスト7

```

100 A=11:B#=STR$(A):C=LEN(B#):PRINT B#,C:PRINT ASC(B#)
110 A=-11:B#=STR$(A):C=LEN(B#):PRINT B#,C:PRINT ASC(B#)
200 A$=" 11":B=VAL(A$):PRINT B
210 A$="11":B=VAL(A$):PRINT B
220 A$="-11":B=VAL(A$):PRINT B
230 A$="A11":B=VAL(A$):PRINT B
240 A$="1A1":B=VAL(A$):PRINT B
250 A$="11A":B=VAL(A$):PRINT B

```

### STR\$



今度は、数字を文字に、文字を数字にしたいときの命令だよ。

「STR\$」というのは、うしろの「( )」の中に、数字や数値変数を入れると、そのまま文字列に直してくれる関数なんだ。

ところが、100行を実行するとわかるように、「11」という数字を文字列にすると、その文字列の長さは「3」になってしまったね。そして、その文字列「B\$」の「ASC」をとると「ASC(X\$)」とするとX\$の先頭の文字のナードカンバーを教えてください。ね。「32」になっている。これは、110行を実行してみるとわかる。

前にも言ったとおり、数を画面へ表示するとき、その数が「+」（プラス）だったら、「+」のかわりに、「」（空白=スペース）を書くんだ。だから文字列に直したときも、そのスペースを先頭に3文字の文字列にしたわけなんだ。当然、その先頭のASCコードは空白を意味する「32」になるよね。

今まで、

```
A=1:PRINT "A=" ; A
```

```
A= 1
```

となっていたね。「=」のうしろにすぐ「1」が表示されなかっただろう。これを何とかしたいな、と思ったら、


```
A=1:A$=STR$(A):PRINT " A=" ;
```

```
LEFT$(A$,LEN(A$)-1)
```

とすると、ちゃんと「A=1」と表示されるんだ。

「VAL」というのは、その逆で、文字列を数値に直す命令なんだ。

# &H HEX\$



文字列はスペース、「-」，数字だけでできていなければならない。  
もし文字があれば，それまでに出てきた数字だけを数値に直す。  
230, 240, 250 行は，それぞれ「0」，「1」，「11」になるだろう。  
(注) 最初の文字が「&H」の場合，16進数といって数値に直すことができるが，数字を16進数としてあつかう，「HEX\$」と一緒にジャンプ編で説明する。)

もちろん，文字列が「698374」のように，あつかえる数値の制限をこえてしまうと，エラーになっちゃうよ。


今度は画面の読みとり方だ。カーソルは，X座標とY座標で決まる。いいかえれば，コンピューターの中に，いつも「X=」，「Y=

リスト  
8

```

100 CLS
110 LOCATE 5,1
120 X1=POS(0):Y1=CSRLIN
130 PRINT "TEST"
140 X2=POS(0):Y2=CSRLIN
200 A$=SCR$(5,1)
210 B$=SCR$(5,1,0)
220 C$=SCR$(5,1,1)
250 COLOR 5,1,3
260 D$=SCR$(5,1,1)
300 LOCATE0,4
310 PRINT "X1=";X1,"Y1=";Y1
320 PRINT "X2=";X2,"Y2=";Y2
330 PRINT A$,B$
340 PRINT "COLOR1=";ASC(C$)
350 PRINT "COLOR2=";ASC(D$)

```



POS(0)  
SCRLIN

」という数値が記憶されていて，「PRINT」命令がくると，記憶しているX，Yの位置から表示するわけなんだ。その位置を変える命令が「LOCATE」だったよね。

「POS」(0)と「CSRLIN」は，そのXとY座標がどこにあるか





POS(0)→4 SCR\$(9.7.0)→"R"  
CSRLIN→11 SCR\$(9.7.1)→2

を<sup>おし</sup>教えてもら<sup>めいれい</sup>う命令<sup>ぎよう</sup>なんだ。100行<sup>ぎよう</sup>から140行<sup>ぎよう</sup>で、変<sup>へん</sup>化<sup>か</sup>をたしかめ<sup>ら</sup>れるよ。


「POS (0) の「0」は、ダミー<sup>で</sup>といって、制限<sup>せいげん</sup>以下<sup>い</sup>の数字<sup>かず</sup>なら何<sup>なん</sup>でもいいんだ。あとで出てくる「MOVE」のPOSITION」という命<sup>めい</sup>令<sup>れい</sup>とは間<sup>ま</sup>違<sup>ちが</sup>えないようにね！

次<sup>つぎ</sup>の200行<sup>ぎよう</sup>から出てくる「SCR\$」は画面<sup>がめん</sup>のどこに、どんなキャラ<sup>きゃら</sup>クター<sup>か</sup>が書<sup>か</sup>かれてい<sup>おし</sup>るか<sup>かんすう</sup>を教<sup>じ</sup>えてくれる関<sup>かん</sup>数<sup>すう</sup>なんだ。実<sup>じつ</sup>は200行<sup>ぎよう</sup>と210行<sup>ぎよう</sup>は、ま<sup>おな</sup>ったく同<sup>おな</sup>じ働<sup>はたら</sup>き<sup>き</sup>をしてるんだよ。「( )」の中<sup>なか</sup>の一<sup>いち</sup>番<sup>ばん</sup>最<sup>さい</sup>後<sup>ご</sup>が「0」のとき<sup>とき</sup>は、省<sup>しょう</sup>略<sup>りやく</sup>して<sup>し</sup>もい<sup>い</sup>い、とい<sup>い</sup>うわ<sup>わ</sup>けだ。この省<sup>しょう</sup>略<sup>りやく</sup>、あ<sup>あ</sup>る<sup>る</sup>い<sup>い</sup>は「0」のとき<sup>とき</sup>は、その場<sup>ば</sup>所<sup>しょ</sup>に<sup>も</sup>あ<sup>あ</sup>る<sup>る</sup>文<sup>もん</sup>字<sup>じ</sup>を教<sup>おし</sup>えて<sup>ぎやく</sup>くれる。逆<sup>さか</sup>に、最<sup>さい</sup>後<sup>ご</sup>の数字<sup>すうじ</sup>が「1」のとき<sup>とき</sup>は、そのキ<sup>おし</sup>ャ<sup>おし</sup>ラ<sup>おし</sup>ク<sup>おし</sup>ター<sup>おし</sup>のカラー<sup>おし</sup>を教<sup>おし</sup>えて<sup>おし</sup>くれるんだ。

SCR\$

SCR\$ (X座標<sup>ざひよう</sup>, Y座標<sup>ざひよう</sup>, 色<sup>いろ</sup> = 1・文<sup>もん</sup>字<sup>じ</sup> = 0)

というしくみだ。



カラーは当然<sup>とうぜん</sup>0-3の間<sup>あいだ</sup>しかなく、CHR\$(0)やCHR\$(3)は画面<sup>がめん</sup>に出<sup>で</sup>ないから、カラー情<sup>じ</sup>報<sup>ほう</sup>を知<sup>し</sup>りたいときは、340~350行<sup>ぎよう</sup>のよ<sup>よ</sup>うに「ASC」で聞<sup>き</sup>かなければい<sup>い</sup>け<sup>い</sup>ない。


当然<sup>とうぜん</sup>「( )」の中<sup>なか</sup>の最<sup>さい</sup>初<sup>しよ</sup>の2つ<sup>すうち</sup>の数<sup>ざひよう</sup>値<sup>ざひよう</sup>, X座標<sup>ざひよう</sup>・Y座標<sup>ざひよう</sup>は「60,

40」のように、テレビの画面のわくをこえちゃうとエラーになって  
しまうんだ。

この「SCR\$」は迷路ゲームなどによく使うから大事な命令だぞ。  
ふつうのパソコンには三角関数 (SIN, COS) や対数関数 (LOG)  
とか、いろんな関数があるけど、ファミコンは数学の研究機械じゃ

●  
リ  
ス  
ト  
9

```
100 A=10:B=3:C=0:D=-10
110 E=A/B:PRINT E
120 E=A MOD B:PRINT E
130 E=ABS(A):PRINT E
140 E=ABS(C):PRINT E
150 E=ABS(D):PRINT E
160 E=SGN(A):PRINT E
170 E=SGN(C):PRINT E
180 E=SGN(D):PRINT E
```



MOD


ないので、バツサリ、カットしてあるんだ。

「MOD」というのは、割り算の余りを出す命令だったね。小数を  
使わない割り算は、

$$10 \div 3 = \text{商 } 3 \cdots \cdots \text{余り } 1$$

としたよね。この「商」を求めるのが「 $10/3$ 」そして、余りを求  
めるのが「 $10 \text{ MOD } 3$ 」というわけだ。

ABS



「ABS」という関数は、うしろの「( )」の中の数値の絶対値を求  
める命令だ。もう学校で「 $|-10|$ 」なんていうのは習ったかな？

絶対値というのは、「+」とか「-」の記号とかをとった数値なん  
だ。この関数がないと、図5 のような時、敵とのきよりを計算す  
るのに、

●図5

X 敵	A 主人公	X 敵
2	5	1
1		9



$$\text{きより} = X - A$$

とすると、敵<sup>てき</sup>とのきよりが「-4」なんて変<sup>へん</sup>なことになっちゃう。  
しかたないから、


$$\text{きより} = X - A : \text{IF } \text{きより} < 0 \text{ THEN } \text{きより} = A - X$$

なんてしなきゃならない。こんなとき、

$$\text{きより} = \text{ABS}(X - A)$$

とすれば一発<sup>いっぱつ</sup>だ!

## SGN



「SGN」というのは、「( )」の中の数値<sup>なか すうち</sup>が「+」か「-」か「0」かを<sup>おし</sup>教えてくれる。「+」だったら「1」,「0」だったら「0」,「-」だったら「-1」だ。

主人公<sup>しゅじんこう</sup>が敵<sup>てき</sup>を追いかけていくゲーム<sup>つく</sup>を作るとき、

$$\text{IF } X > A \text{ THEN } A = A + 1 \quad (\text{右へゆく})$$

$$\text{IF } X = A \text{ THEN } A = A \quad (\text{そのまま})$$

$$\text{IF } X < A \text{ THEN } A = A - 1 \quad (\text{左へゆく})$$

とするよりも、

$$A = A + \text{SGN}(X - A)$$

とすると一発<sup>いっぱつ</sup>でOKだ! わかるかな?

もう一つ、「A」、「B」という二つの変数の中身を交換するには

$$C = A : A = B : B = C$$

のようになさなければいけないのだが、簡単に

$$\text{SWAP } A, B$$

でできる。「SWAP」はうしろの二つの変数の値を交換しろという命令で、もちろん、文字変数周<sup>しゅう</sup>士にも使える。

プログラムの途中でコメントを入<sup>い</sup>れたい時は

10 ' TEST

20 REM MAIN

のように「'」や「REM」と書くと、そこからうしろの部分は命令として見えないので便利だ。

さて、もうこれで、ベーシック<sup>きほんてき</sup>の基本的な命令<sup>めいれい</sup>は、ほとんどすんじやった。意外<sup>いがい</sup>にやさしかっただろう? 次<sup>つぎ</sup>はいよいよ、グラフィック・ワールドへ出<sup>しゅつ</sup>発<sup>ぱつ</sup>だ!

# 作戦指令 006


## テレビへ落書き 3つの方法





# 006-1

## PRINT文にも いろいろあるぞ



らくが 落書きなんて、とんでもない！ごめん、ごめん。

ちよっとむずかしい変数や文法なんて忘れて、これから少し、絵を書く勉強をしてみようか。

いちばんかんたん 一番簡単なのは、PRINT 命令だね。ところで、「LOCATE」と「PRINT」の関係はもう頭に入ってるかな？

1Ø CLS

2Ø PRINT "TEST"

とすると、画面の左上すみに「TEST」と出るはずだね。コンピューターは「PRINT」命令に出会ったとき、どこから文字を書いたらいいか、をおぼえている。その数値を知りたいときは「POS (Ø)」と「CSRLIN」という命令で教えてくれるはずだね。



1 Ø CLS

2 Ø X1=POS (Ø) : Y1=CSRLIN

3 Ø PRINT "TEST"

4 Ø X2=POS (Ø) : Y2=CSRLIN

5 Ø PRINT X1, Y1, X2, Y2


最初「X=Ø, Y=Ø」の位置にあったカーソルの位置が「TEST」と表示したあと「X=Ø, Y=1」になっているの<sup>い</sup>ち<sup>ち</sup>がわかるよね。

ところで、50行の「,」は何<sup>なん</sup>だ<sup>だ</sup>ろうか? 30行を、

3 Ø PRINT "TEST",

としてみよう。

今度は「TEST」と書かれた同じ行に、



X1, Y1, X2, が表示されて、しかも Y2 の値は同じで、X2 の値だけが「7」に変化しているね。

なぜかという<sup>めい</sup>れい<sup>ひ</sup>ょうじ<sup>じ</sup>、「PRINT」命令で表示させる文字や数字の一番うしろに、何にもないときは、カーソルの位置は次の行の一番左 (X2=Ø, Y2=Y1+1) にうつる。それに対して「,」が最後に書かれているときは、Y2 が変化しないで、X2 が変化するんだ。

ファミコンのキャラクター画面は、横28字 (X=Ø~27) だったね。「PRINT」文の最後に「,」か「;」がないときは、「改行」といって、次の「PRINT」命令で何かを表示するときは、次の行から書きはじめなさい、という意味になるんだ。

3 Ø PRINT "TEST";


5 Ø PRINT X1; Y1;

X2; Y2


としてみよう。今度は同じ行につづけて、

TEST Ø Ø 4 Ø

と表示される。それぞれの数字の前に一字空白 (スペース) があるのは、数値を表示する時の約束で「+」の意味だったよね。








;

,



つまり、「;」があるときは、カーソルの位置を変えない（改行しない）で、続けて書け、という意味なんだ。

「,」があると、X2は、

$$X2 = X1 + (A / 7 + 1) * 7$$

直前に表示されるものが文字列 (A\$) のときは、

$$A = \text{LEN}(A\$)$$

数字 (X) のときは、

$$A = \text{LEN}(\text{STR}\$(X))$$

$$(\text{=その数字}(X)\text{のケタ数} + 1)$$

であらわされる。つまり、横28文字を7ずつの4つのブロックに分けて、その前に表示されたものと最低1つの空白をあけて表示しろ、ということになるんだ。

30 PRINT "7WORDS7",

30 PRINT "6WORDS",

など、30行の「PRINT」文のあとの文字列の長さをいろいろかえてみると、働きがよくわかるだろう。

10 CLS

20 FOR I=0 TO 15

30 FOR J=0 TO 15

40 LOCATE J, I:PRINT CHR\$(I\*  
6+J);

50 NEXT: NEXT




このプログラムを打ち込で<sup>う</sup>CON<sup>こん</sup>しながら、キャラクタコード<sup>ひょう</sup>表<sup>ひょう</sup> Bを見てみよう。CHR\$(0)からCHR\$(31)まで(キャラクタ<sup>ひょう</sup>テーブル Bの A0から D7の 32個)が表示<sup>ひょうじ</sup>されない他は、コード表<sup>ひょう</sup> Bの<sup>とお</sup>ひょうじ<sup>ひょうじ</sup>に表示されるね。(CHR\$(32)は空白<sup>くわくはく</sup>=スペースだ)

コントロール・コード この 0 から 31 は普通<sup>ふつう</sup>、コントロール・コード<sup>おお</sup>といってコンピューターが<sup>そうき</sup>いろいろな操作<sup>そうさ</sup>をやるときに使<sup>つか</sup>っているんだけど、ファミコンではできるだけ多くのキャラクター<sup>おお</sup>を表示<sup>ひょうじ</sup>したいため、この場所にもキャラクター<sup>お</sup>を割<sup>わ</sup>り当<sup>あ</sup>てて 2 重<sup>じゆう</sup>に使<sup>つか</sup>っているんだ。まへの「INKEY\$」でやったように、たとえばカーソルキーの「▼」のコードナンバーが「31」だったね。だから BG 画面<sup>がめん</sup>ではこの 0 から 255 までのキャラクター<sup>ぜんぶつか</sup>が全部<sup>おお</sup>使<sup>つか</sup>えたけど、普通のベーシック画面<sup>がめん</sup>で「PRINT CHR\$( )」として表示<sup>ひょうじ</sup>されるのは、32 から 255 まで、となってるんだ。

じゃあ、キャラクターテーブル A のキャラクター<sup>つか</sup>は使<sup>つか</sup>えないのか、という<sup>お</sup>とそうでもない。

ま<sup>まえ</sup>前のプログラムを「RUN」して、各<sup>かく</sup>キャラクター<sup>ひょうじ</sup>が表示<sup>ひょうじ</sup>された所<sup>ところ</sup>





## CGEN




で、ダイレクトに「CGEN 0」としてごらん。画面がむちゃくちゃになったね。でもよく見ると、テーブルBのキャラクターが書かれていた所に、マリオやレディが書かれているよね。

画面全体に書かれている模様はなんだろう？

スペース（空白）も一文字で、コードナンバーは「32」だったのは何回も言ったよね。ところで、テーブルAのコードナンバー「32」はなんだろう？レディの横顔だね。画面一杯にレディの横顔が「スペース」のかわりに表示されているんだ。いくらレディがかわいいといっても、ちょっと気持ち悪いなあ。



このままじゃ、なんにもできないから「コントロールキー」を押しながら「D」のキーを押してごらん。もともどったね。（Ctrl-D というように書いてあると、コントロールキーを押しながらDのキーを押す、という意味だよ。）

## Ctrl-D




# 006-2

## ファミコンは何重人格？




バックドロップ

キャラクタ画面



スプライト面



「CGEN」とはどんな命令だったのか？その前に、ファミコンの顔（画面）は、一体どうなっているんだろう？

アニメ映画を作るとき、普通、「セル」と呼ばれる透明なプラスチックにいろんな絵を書いて、重ね合わせていることは君達も知っているよね。（図1）

①、②、③の重ね合わせで、太陽が山のうしろにかくれたり、汽車の下にある線路が見えなかったりするわけだ。

ファミコンも考え方は同じなんだ。まず、一番下に「バックドロップ」面がある。この画面に文字やキャラクターは書けなくて、何か一色にぬる事ができるわけだ。ベーシック画面ではいつも黒になっているね。背景画面とも呼ぼう。そして「キャラクタ画面」、ファミコンのマニュアルでは「バックグラウンド面」になっているけど、意味からするとこっちの名前が合っていると思うね。つまり文字・数字・キャラクターテーブルBなどを表示する面で、ベーシックのプログラムを使ったりするのもこの面だ。それに「CGEN」命令を使うと、キャラクターテーブルAも使えるそうぞ。

そして問題の「スプライト面」だけけど、このスプライト面は、2枚あるんだ。おもに、キャラクターテーブルAのキャラクターを表示したり、動かしたりする。この2枚はそれぞれ、「スプライト前」と「スプライト後」ともいえるんだ。


つまり、一番下に「バックドロップ」面があり、次に「スプライト後」、「キャラクター」、一番上が「スプライト前」になっているんだ。（図2）

この4枚のうち、1枚だけ使える大きさが違うんだ。そして、「CLS」をしたあと、「CGEN 0」をもう一度してみよう。




● 図 1


A




B





C



C B A



A B C




レディの横顔オンパレードだけど、カーソルは何とか見えるね。  
 カーソルキーでカーソルを移動させてみよう。画面にはレディが出て  
 てるのに、移動できない所があるだろう。移動できる所を数えてみ  
 ると、ちゃんと28×24あるはずだね。(Ctrl-Dで元に戻そう。)

ファミコンの画面は、本当は横32文字、たて30行もあるんだ。た  
 だ、家庭用のテレビは、いろんな会社のがあって、全部をきれいに  
 うつせないかもしれないから、キャラクター面は、そのうち28×



● 図 3




24 (図3) しか使ってないんだ。上下左右に余った分は、スペース(空白)でうめてるのは、さっきの実験でわかるよね。

BG画面はどうなんだろう。ホップ編で練習したように、BG画面でいろんな絵を書いたあと、ベーシック画面に来たら、せっかく書いた絵が消えてしまった。と思ったら「SYSTEM」命令でBG画面に行くとちゃんと残っていたね。

今、ためしに、BG画面でいろんな絵を書いてみよう。(かんたんな絵でいいんだ。)そのあとベーシックにもどってみると、まっくらになってるね。ダイレクトで、

VIEW



VIEW

と打つと、今かいた絵が出てきたね。

CLS



とするか、シフトキーを押しながら、CLR キーを押すと、当然、今の絵は消えてしまう。もう一度、

## VIEW

とすると、また元の絵が出てくるだろう。

つまり、BG 画面でのお絵かきは、画面に表示する「キャラクター」面とは、全然別の所で行なっている。そして出来上がった絵は「VIEW」という命令で、そっくりそのまま「キャラクター」面へ移動(転送という。)されたわけだ。だから「キャラクター」面を「CLS」命令で消してしまっても、もとの所にはその絵が残っているんだから、もう一度「VIEW」命令で「キャラクター」面に同じ絵が出てくるしくみなんだ。(「VIEW」命令は、BG 画面を使うゲームには絶対必要な命令なのだ!)

この BG 画面は横28字、たて21行で、キャラクター面の下3行を残した上の部分へ、そっくりそのまま転送される。

ところで、文字や、グラフィックスのキャラクターは、すべて「8×8」の小さな点によって作られている。というのは覚えてるかな？(ホップ編85ページ)

おもに、マリオやレディなどのグラフィックキャラクターを動かすのに使うスプライト面は、その動きを細かく表現するために、画面に表示する単位を、この小さな点(ドットと呼んだよね。)で示している。このスプライト面の大きさは、バックドロップ面と同じ32×30行だ。1字をドットで表わすと8×8だから、スプライト面をドットで表わすと、

X... 8 × 32 = 256 ドット    X座標... 0 ~ 255

Y... 8 × 30 = 240 ドット    Y座標... 0 ~ 239

ということになるよね。

10 SPRITE ON

20 DEF SPRITE 0, (0, 1, 0, 0, 0  
0) = "DEFG"

30 INPUT "X=", X

40 INPUT "Y=", Y

50 SPRITE 0, X, Y

60 GOTO 30



```

OK
10  SPRITE ON
20  DEF SPRITE 0, (0, 1, 0, 0, 0) =
   "DEFG"
30  INPUT "X=", X
40  INPUT "Y=", Y
50  SPRITE 0, X, Y
60  GOTO 30
RUN
X=232
Y=208
X=■

```

これはアキレスを INPUT した座標<sup>ざひよう ひようじ</sup>に表示するプログラムなんだ。  
 X, Y とも 0~255<sup>あいだ すうじ</sup>の間の数字ならエラーにならず, X=0~240, Y=5~220以外の数値だと, どこかがちよん切れるだろう。また X=16, Y=24でキャラクター面<sup>めん ひだりうえ</sup>の左上すみ, X=232, Y=208で右下すみ<sup>みぎした ひだりうえ</sup>のところに, アキレスの左上<sup>み</sup>がくるのがたしかめられるかな? 図3を見ながら, いろいろとためしてみよう。

0~255のドット座標系<sup>ざひようけい</sup>と「LOCATE」などにつかう 0~27の座標系<sup>ひようけい</sup>と2つも出てきて, 頭<sup>あたま</sup>がおかしくなっちゃうかもしれないけど, 2つの区別<sup>くべつ</sup>は, 必ずつけられるようにしようね。

この2つの関係<sup>かんけい</sup>は,

X 1, Y 1 = キャラクター面<sup>めん</sup>の X, Y 座標<sup>ざひよう</sup>

X, Y = スプライトの X, Y 座標<sup>ざひよう</sup>

とすると,

$$X = (X 1 * 8) + 16$$

$$Y = (Y 1 * 8) + 24$$

になっている。

# 006-3

## スプライトが重なりと



ホップ編で、スプライトが横に4つ以上重なりと表示されない、  
ということにふれたけど、どういう事なんだろう？

```

100 SPRITE ON
110 FOR I=0 TO 4
120 DEF SPRITE I, (0,1,0,0,0,
    0 )="DEFG"
130 SPRITE I,100,I*20+10
140 NEXT
    
```



アキレスが、たてに5つ表示されたね。今度は、

```

130 SPRITE I, I*30,100
    
```

にしてみよう。4つしか出てこないね。





130 SPRITE 1, 1\*30, 100+1\*2

とすると、<sup>すこ</sup>少しづつ<sup>たか</sup>高さがずれながら4つのアキレス<sup>ひょうじ</sup>が表示された  
あと、<sup>あたま</sup>頭の<sup>か</sup>欠けた<sup>め</sup>5つ目のアキレスが出てきたね。

キャラクターの<sup>ところ</sup>所で、ファミコンはセルを<sup>かき</sup>重ねて<sup>か</sup>書いているみた  
いなもんだ、ということ<sup>い</sup>を言ったよね。ところで、このスプライト

面にはセルを使う制限があるんだ。たとえば、アキレスやマリオは、たて16、横16のドットで表わされてることは前にも言ったね。このドット1つが1cm×1cmもある大きなものとしよう。

図4の④のような、横に長く、厚さも1cmあるすごく厚いセル(棒みたいなものだ。)のまん中あたりに16cmのキャラクターがある。つまり「DEF SPRITE」なんかで、決まったキャラクターは、この④がたて16本ならんだものと考えられるね。とすると、プログラム①の場合は、図4-③のようになる。それに対し、②にすると、⑤のようになるのがわかるね。横から見たのが、それぞれ⑩と⑪だ。

ではプログラム③はどうだろう。図4-⑥になり、横から見たら5⑤になるのがわかるかな。

このスプライト面に許されているセルの厚みは4cm、つまり4枚だけなんだ。そこでプログラム②、図4-⑦では4枚目までしか表示されなくて、③、図4-⑧では5枚目(1番上)の絵は頭(図では左)の①の部分が厚さ5cmで表示されないけど、残りは表示された、というわけなんだ。

スプライトを使ってゲームを作るときは、この点に注意しなきゃいけないよ。

ところで、さっきから、

DEF SPRITE ~ = "DEFG"

なんて使ってきたけど、これはどういうことなんだろう？

アキレスを表示するには、たとえば、


DEF SPRITE ~ = CHR\$(64) + CHR\$(65) + CHR\$(66) + CHR\$(67)

のはずだったよね。ところで、CHR\$(64)から(67)は、それぞれ、キャラクタテーブルBでは「D」から「G」だったね。そこでメモリーや入力をやさしくするために、使ったわけだ。

このように、キャラクタテーブルAとBとは、コードナンバーは同じでも、その数値がどちらのテーブルを使うかさえ決めればいわけなんだ。




電源を入れた状態では、キャラクタ面にはテーブルB、スプライト面にはテーブルAを使うと決まっているんだけど、それを君達が

4枚目まで







● 図 4




● 図 4


E



F




G





## CGEN



かえることもできるんだよ。

「CGEN」という命令<sup>めいれい</sup>がそれなんだ。(やっと出てきたね。)

キャラクター面<sup>めん</sup> スプライト面<sup>めん</sup>

CGEN	0	テーブルA	テーブルA
CGEN	1	テーブルA	テーブルB
CGEN	2	テーブルB	テーブルA
CGEN	3	テーブルB	テーブルB

というのが「CGEN」と、そのあとに続く数値<sup>すうち</sup>の意味なんだ。

電源<sup>でんげん</sup>を入れた直後は、「CGEN 2」になっているわけだね。そして「Ctrl-D」というのは、「CGEN 2」の命令<sup>めいれい</sup>だったとも言えるよね。

この「Ctrl-D」。他にもいろんな働き<sup>はたら</sup>を一度<sup>いちど</sup>にしてくれるんだ。


「SPRITE ON」でスプライト面<sup>めん</sup>を一度表示<sup>ひょうじ</sup>すると、「CLS」命令<sup>めい</sup>では、スプライト面<sup>めん</sup>の表示は消えなかったね。「CLS」は、キャラクター面<sup>めん</sup>を消すだけの働き<sup>はたら</sup>をするからなんだ。スプライト面<sup>めん</sup>を消すには、「SPRITE OFF」という命令<sup>めいれい</sup>しかなかったね。

ところが、この「Ctrl-D」、さっきの「CGEN 2」とこの「SPRITE OFF」、そしてジャンプ<sup>へん</sup>編<sup>なら</sup>で習う、色の組み合わせを変えたものを元<sup>もと</sup>にもどす、という働き<sup>はたら</sup>を一度<sup>いちど</sup>に行なってしまうんだ。

「Ctrl-」, つまり、「CTR」キーを押しながら、他のキーを押す、というのには、いろんな種類<sup>しゆるい</sup>があって、とても便利<sup>べんり</sup>なんだ。くわしい説明<sup>せつめい</sup>は巻末<sup>かんまつ</sup>の作戦資料<sup>さくせん しりょう</sup> 103を見てほしい。

# 006-4

## スプライトを動かすには



ホップ<sup>へん</sup>編では、「スプライト<sup>めいれい</sup>命令<sup>つか</sup>」を使って、スターシップとスターキラーを動かしたよね。ところが、キャラクターテーブルAを見てみよう。マリオだけで7種類<sup>しゅるい</sup>もあるね。「ドンキーコング」や「マリオブラザーズ」なんかを見ると、マリオがチョコチョコと手足<sup>てあし</sup>を動かしながら歩いている。この感じを出すには、キャラクターテーブルAのマリオの「WALK1」,「WALK2」,「WALK3」を順番<sup>じゆんばん</sup>に少しずつ<sup>うごく</sup>動かしながら表示<sup>ひょうじ</sup>しなければいけないんだ。

●  
リスト1

```

100 CLS
110 SPRITE ON
120 FOR I=0 TO 2
130 DEF SPRITE I, (0,1,0,0,0)=CHR$(I*4)
+CHR$(I*4+1)+CHR$(I*4+2)+CHR$(I*4+3)
140 NEXT
150 FOR I=200 TO 0 STEP -1
160 SPRITE I MOD 3
170 SPRITE (I+1) MOD 3
180 SPRITE (I+2) MOD 3,I,100
190 PAUSE 3
200 NEXT
    
```

これが、マリオを X=200, Y=100 の位置<sup>いち</sup>から移動<sup>いどう</sup>するプログラムなんだ。120 行<sup>ぎよう</sup>から140 行<sup>ぎよう</sup>でマリオの歩く<sup>ある</sup>3つのポーズを、それぞれ「SPRITE」の0から2に定義<sup>ていぎ</sup>しているのがわかるね。



150 行から200 行でマリオを移動している。

1 の値 <small>あたい</small>	消すスプライト <small>け</small>	描くスプライト <small>えが</small>
2 0 0	2 , ( 0 )	1
1 9 9	1 , ( 2 )	0
1 9 8	0 , ( 1 )	2
1 9 7	2 , ( 0 )	1
1 9 6	1 , ( 2 )	0
}	}	}
	↓ ↓ ↓	
	160行 170行	180行

3つのスプライトが重ならないように、まず他の2つを消したあと、使うスプライトを描くために「MOD」を使っている。(170 行

リスト2

```

100 CLS
110 SPRITE ON
120 FOR I=200 TO 0 STEP -1
130 J=I MOD 3
140 DEF SPRITE 0,(0,1,0,0,0)=CHR$(J*
4)+CHR$(J*4+1)+CHR$(J*4+2)+CHR$(J*4+3)
150 SPRITE 0,I,100
160 PAUSE 3
170 NEXT

```

は、なくてもよいが、わかりやすくするために、付け加えた。

また、次の方法もある。

これは「SPRITE 0」しか使わずに、その中身を、そのたびに書きかえる。という方法だ。

ところが、どちらの方法を使っても問題は残る。マリオは必ずしも左向きだけ、とは限らないんだ。ホップ編を思い出そう。マリオを右向きにするには、「CHR\$」の並べ方を変えた上に、「DEF SPRITE」の「( )」の中に4番目の数字(4番目のパラメーターと

いう。)を1にしなきゃいけない。その上、いろんなキャラクターを  
 ひょうじ 表示していると、メモリーがたりないし、第一、「DEF SPRITE」

リスト  
3

```
100 CLS
110 SPRITE ON
120 DEF MOVE(0)=SPRITE(0,7,1,1,0)
130 POSITION 0,200,100
140 FOR I=0 TO 100
150 MOVE 0
160 IF MOVE(0)=-1 THEN 160
170 NEXT
```

は0から7までの8つしか定義できない。

そこで、新しい命令だ。

あたらしい命令が、いくつか出てきたね。

DEF MOVE( )=SPRITE( )

MOVE

MOVE( )

POSITION

そして、あとで出てくる、

CUT, ERA, XPOS, YPOS

と、Ver.3で使える、

CAN, CRASH, VCT

をふくめて、「MOVE=ムーブ」系のコマンド(命令)群とよぼう。

それに対し、今まで使ってきた、

DEF SPRITE n, ( )=文字列

SPRITE n, X, Y

SPRITE ON


SPRITE OFF

は、「SPRITE」系の命令群といえる。


つまり、「スプライト」面にキャラクターを表示するには「MOVE」

系と「SPRITE」系の2つの流れがあるわけだ。

この「MOVE」系の命令は「SPRITE」系の命令をより進めたも



MOVE系



スプライト系



ので、他のパソコンにはない強力な命令になっているんだ。


君達も、どこかの雑誌で読んだかもしれないけど、一つのキャラクターを描いて、それを移動させるためには、かなり多くのプログラムとメモリーを使うんだ。ところがファミコンには、わずか1982バイトしかメモリーがない。そこで普通のパソコンのように絵を書くための命令は、全部なくして、ファミコン内部に用意してあるキャラクター(テーブル A, B)を自由に使えるようにした、というわけなんだ。

ファミコンとパソコンで一番違うのは、この「MOVE」系の命令群にある、って言っても間違いじゃないんだ。

その他では、パソコンの持っている機能をけずって小さくしているファミコンだけど、これだけは、兄貴分のパソコンよりもずっと進んでいるんだ。(何しろ「SPRITE」系の機能でさえ、MSX や X-1 など、一部のパソコンしか持っていないんだから。)

# 006-5


## MOVEというのは 動くという意味



さて、いよいよ「MOVE」の脱明だよ。

DEF MOVE (n) = SPRITE (A, B, C, D,  
E, F)

DEF MOVE  
MOVE



この命令系の特徴は、「DEF MOVE」であるキャラクターとその動きを定義しておく、と、「MOVE」という命令で自動的に定義された動きを行なってくれる、という点だ。しかも、「DEF SPRITE」のように、キャラクターが右向きとか、ななめとかで、その並び方を変えたりする必要がなく、キャラクターの名前だけで、どの「CHR\$」をどういうふうに組みあわすかを自動的に決めてくれるんだ。

「DEF MOVE」も「DEF SPRITE」のように、0から7までの8個を登録できる。そのナンバーが、最初の「( )」の中に入る。つまり、「n」は0から7の間の数字が入るんだ。(DEF SPRITE は、このナンバーだけが逆に「( )」の外だったね。)

次に「=」の右側の「( )」の中だ。「A」には0~15の数が入る。

これはキャラクターの種類だよ。

0 = マリオ	8 = スターキラー
1 = レディ	9 = スターシップ
2 = ファイターフライ	10 = 爆発
3 = アキレス	11 = ニタニタ
4 = ペンペン	12 = レーザー
5 = ファイアーボール	13 = シェルクリーパー (カメ)
6 = 車	14 = サイドステッパー (カニ)
7 = スピナー	15 = ニットピッカー (トリ)



「B」は動きの方向だ。

8 1 2

↖ ↑ ↗


7 ← ∅ → 3

∅ ~ 8 が入る

↙ ↓ ↘

6 5 4

∅ で停止



「C」は移動の速さだ。60ドットを移動するのに、何秒かけるかを表わすんだ。1~255までの数字を入れる。1だと60ドットを1秒という意味だ。(∅を入れると256を入れたと同じ、つまり一番遅くなる。)

「D」は「MOVE」命令を一回出したとき、何ドット移動するか、移動量を入れるんだ。1~255の数値を入れる。入れた数値の2倍のドットを移動するから、∅を入れると表示されないよ。

「E」は0か1を入れる。0だと、このキャラクターは「スプライト前」面、1だと「スプライト後」面に書かれるんだ。

「F」はカラーのパレットナンバーで、0~3の数字が入るんだ。

「E」と「F」は省略して、

**DEF MOVE (0)=SPRITE (0, 1, 1, 5)**

のように書くこともできるんだ。

こうやって定義したキャラクターを、一番最初どこに表示するか、スタート地点を決める命令が「POSITION」命令なんだ。

**POSITION 0, 100, 50**

というのは、「DEF MOVE」命令で定義したナンバー「0」のキャラクターは「X=100, Y=50」の地点からスタートしたよ。という意味なんだ。

さあ、これですべてを定義した。いよいよ動かすぞ！


**MOVE 0**

で、このキャラクターは、最初に定義した動きを一回だけ行なう。


**DEF MOVE (0)=SPRITE (0, 1, 2, 10)**

と定義していれば、1秒間30ドット(60ドットを2秒だから)の速度で10ドットだけ(つまり1/3秒間)上へ移動するわけだ。


この「MOVE」、動かしたりキャラクターがいくつもあるときは、



POSITION



MOVE





## MOVE 0, 1, 4, 5

のように、一度につづけて書くこともできるんだ。

10 SPRITE ON:CLS

20 DEF MOVE(0)=SPRITE(0, 3, 3  
, 255)

30 POSITION 0, 0, 100

40 MOVE 0

RUNすると「OK」の表示が出ていのに、マリオが<sup>はし</sup>どんどん走っている。「LIST」をとったり、コンピューターに他の仕事をやら<sup>ほか しごと</sup>しても、マリオは動いている。マリオは画面を2回横切ってや<sup>が めん かいよこ ぎ</sup>って停止した。

これは、20行でマリオの1回の動きで「255」つまり2倍の510ドット移動すると決めていたので、「動け=MOVE」と命令したら、510ドット移動し終るまでマリオは意地になって(?)動いているんだ。ところがコンピューターの方は「MOVE」って命令を出したら、もう自分の仕事はすんだから、他の仕事ができるようになったんだ。

あやつり人形の使い手が8人いて、コンピューターは舞台監督。  
「DEF MOVE」で人形使いの一人一人にどの人形を、どうい<sup>にんぎようつか ひとりひとり にんぎよう</sup>  
う風に動かすかを指令していると思えばいいね。「ハイ、0番さん」と<sup>ふう うご し れい おも</sup>  
いうと、0番の人形使いは約束通り、マリオ人形を動かしているわ<sup>ばん にんぎようつか やくそくどう にんぎよう うご</sup>



けだね。

## 6 Ø MOVE Ø

を追加しよう。

「MOVE」命令一回で、<sup>めいれい</sup> Mario は画面を 2 回横切ったから、<sup>めいれい</sup> 命令を 2 回にしたら、4 回横切るだろう？


残念でした！この人形使いは、仕事熱心だから、1 回命令をもらって実行中（Mario を移動させている間）は、次の命令が聞こえなくなってるんだ。だから、Mario は 2 回しか画面を横切らない。

どうしても 4 回横切らすには、1 回目の「MOVE」命令をやり終えるまで時間を待ったあと、2 回目の「MOVE」命令を出さなきゃいけない。時間待ちには「PAUSE」だったから、

## 5 Ø PAUSE 25 Ø Ø

ところが、今度は、1 回目の命令のあと、しばらく止まっている。

「PAUSE」のあと数値が長すぎたんだ。そこで登場するのが「MOVE ( )」だ。



MOVE ( )

## 5 Ø IF MOVE (Ø) <> Ø THEN 5 Ø

これは、もし「MOVE」命令で動きはじめたキャラクターがまだ動いていたなら、50 行をもう一度、つまりこのまま待っていて、動き終えたら次の行へ行きなさいよ、という命令になってるんだ。

この「( )」の中の数字は「DEF MOVE」や「MOVE」で指定したキャラクターナンバーだ。

## 5 Ø PRINT MOVE (Ø)

## 55 GOTO 5 Ø

と直して「RUN」してみよう。キャラクターが動いている間は、ずっと「-1」で、動き終えて止まったとたんに、「Ø」になるのがわかるね。つまり、「MOVE ( )」は、そのナンバーのキャラクターが移動しているかどうかを知る関数だったんだ。


55 行を消して、もう一度、

## 5 Ø IF MOVE (Ø) <> Ø THEN 5 Ø

## 6 Ø MOVE Ø



としてみよう。

今度はちゃんと続けて 4 回 Mario が画面を移動したね。1 回目の命令を終えるまで待ってから、2 回目の命令を伝えただから、人



XPOS ( )

YPOS ( )



CUT

形 使いナンバー「0」も、今度は聞こえたわけだね。

50 PRINT XPOS (0), YPOS (0)

60 GOTO 50

としよう。画面には次々と数字が出てくるね。

この「XPOS ( )」と「YPOS ( )」は、それぞれ「( )」の中のナンバーのキャラクターの、現在の X 座標、Y 座標を教えてくれる関数なんだ。カーソルキーの X 座標を教えてくれる「POS」と違って、～255のどれかの数字が出てくる、というのはわかるよね。

じゃあ、これを使ってキャラクターが、ある場所へきたら止めることはできるんだろうか？

50 IF XPOS (0) > 100 THEN CUT

0: GOTO 70


60 GOTO 50

70 PAUSE

これで、OKだ！マリオが画面まん中近くで止まっただろう。「CUT」というのは、動いているキャラクターを、その場所で停止させる命令なんだ。XPOS (0) が100をこえると停止させて70行へいく。70行では「PAUSE」でうしろに数字がないから、何かキー入力があるまでは、待っている、ということなんだ。



## ERA



ここで何かキーを押すと、またマリオは動き出し、残っていた画面横断、一回半をやりとげる。

50 行の「CUT 0」を「ERA 0」に変えてみよう。

今度は、マリオが消えちゃった。キーを押してもプログラム終了の「OK」が出ただけで、マリオは消えたまままだ。この「EAR」は、動いているものも、止まっているものも、全部消しちゃう働きをするんだ。

ダイレクトに、

### MOVE 0

を実行してみよう。さっき消えた所から、マリオが出てきて、やはりさっき中断した動作の残り、一回半、画面横断をやっただろう。

ここで注意しなくちゃいけないのは、「CUT」にしろ、「ERA」にしろ、あくまで最初の命令の動作を中止しただけで、そのあとの「MOVE」命令は、残りの部分を再開しろということにしかならな

いんだ。

マリオが、画面中央まできたら、一度停止したあと何かのキーを押したら、新しく動作を始める（画面中央から2回画面を横切るためには、

●  
リス  
スト  
4

```
10 CLS:SPRITE ON
20 DEF MOVE(0)=SPRITE(0,3,3,255)
30 POSITION 0,0,100
40 MOVE 0
50 IF XPOS(0)>100 THEN 100
60 GOTO 50
100 CUT 0
110 BEEP:PAUSE
200 MOVE 0
```

ではだめで、

```

10 CLS:SPRITE ON
20 DEF MOVE(0)=SPRITE(0,3,3,255)
30 POSITION 0,0,100
40 MOVE 0
50 IF XPOS(0)>100 THEN 100
60 GOTO 50
100 CUT 0
110 BEEP:PAUSE
200 DEF MOVE(0)=SPRITE(0,3,3,255)
210 MOVE 0

```

のように「DEF MOVE」で再定義しなけりやいけないんだ。この場合は「POSITION」命令でスタート位置を指定しなくても、前のX, Y座標が残っているから、元の位置から動きはじめるんだ。一度定義したキャラクターナンバーのX, Y座標は、「POSITION」命令 (Ver. 3の「CAN」か) で指定しない限り、いつまでも残っているんだ。

「DET MOVE」でキャラクターを動かす人形使いさん、ガンコでゆうづうが聞かない分、記憶力はバツグンだね！

ところで、この「CUT」、「ERA」とともに「MOVE」と同じく、


**CUT 0, 1, 5**

のように、いくつものキャラクターに一度に命令が出せるのも覚えておこう。

さて、いよいよ、ゲーム作りに挑戦だ！

# 作戦指令 007


## おっかけゲームに 挑戦だ





# 007-1

## どんなゲームを作ろうか



前に作ったゲームは敵との撃ち合いだったね。今回はどんなゲームにしようか？

BG画面を使って、道を書いて、主人公が逃げまわる。敵はファイターフライにしよう。ファイターフライが、主人公ペンペンを追いかけてまわすことにしよう。

ただ逃げるだけじゃつまんない。歩いていった道にはマークをつけて、全部の道を歩き終わったら、その面は終わり、というようにしよう。

一面が終わると次の面に進むんだけど、そのたびにBG画面のデーターをLOADするのも時間がかかるね。

それじゃあ、一面終えて、レベルが上がるたびに、1個ずつカベをふやして通れない所を多くすると、むずかしくなるわけだ。しかも、ふえた分のカベは、ペンペンは通れないけど、ファイターフライは通れてしまうんだ。


まず元になる迷路をかいてみよう。

### SYSTEM (Ver.3ならBGTOOL)

でBG画面を呼び出そう。使うキャラクターはテーブルBの「I6」、「I7」、「J0」、「J0」、「J1」、「J2」、「J3」、「J4」、「J5」、「J6」、「J7」だ。コード表Bでいうと「CHR\$(222)」から「CHR\$(231)」までを使って書いてみる。画面をこしたかどうかの判定も一諸にやりたいので、まわりもすべて、このキャラクターを使って囲んでしまう。


とりあえず、できたのが図1だ。もちろん、この面は自分でいろいろ作ってみるのもいいだろう。

今度のゲームは、いろんな面を作っても楽しめるのでテープへの



BG画面へ

● 义 1

[illegible]

SAVEは、プログラムのあとにしよう。でも、電源が抜けたりすると大変だから、とりあえずSAVEしちやおう。SAVEのしかたは、もう知ってるよね。

「ESC」キーを押したあと「STOP」キーを押してプログラムモードへ切りかえよう。

こんかい　　ふん　　つか  
 今回は、「DEF　MOVE」文を使うけど、  
 ペンペンもファイターフライもそれぞれ、上  
 げき　きゆう　ほうこう　　つか  
 下左右4方向、合わせて8つも使う。これを  
 ふつ　　ていぎ　　めもり　　つか  
 普通に定義していくとメモリーをたくさん使

ってしまう。そこで前<sup>まえ</sup>に勉強<sup>べんきょう</sup>した、サブルーチン<sup>つか</sup>を使おう。

```
5000 DEF MOVE(N)=SPRITE(N,V
      ,1,4):MOVE N:RETURN
```



つまり、ペンペンとファイターフライがそれぞれ「DEF  
MOVE」のキャラクターナンバー 2, 4 であるのを利用して、「DEF  
MOVE」の 0 から 7 のうち 2 と 4 だけを使う。「N」に 2 か 4 を入  
れて、「V」に方向を入れて「GOSUB 5000」とすれば、ペンペン  
か、ファイターフライが「V」の方向に動く、というサブルーチン  
だ。

100 SPRITE ON:CLS:VIEW

110 SC=0:SE=406

これはゲーム自体の初期設定だから、一番頭に置く。「VIEW」は  
BG 画面をキャラクター画面に転送する命令だったね。SE は空白の  
場所の数だ。

```
200 FOR I=1 TO 27:FOR J=1 T  
0 19:IF SCR$(I,J)=CHR$(  
215) THEN LOCATE I,J:PR  
INT " ";
```

CHR\$(215) はリングだね。ペンペンが歩いていったあとには、  
このマークを置くことにする。

一面クリアしたあとは、このリングマークを一度消さなきゃい  
けないね。それが 200 行だ。「FOR」が 2 つあるから、

210 NEXT:NEXT:M=0

次に乱数で新しいカベを発生させている。230 行でカベの X 座  
標、Y 座標を出す。その位置に何も置かれてなければ、そこへ新し  
いカベ「CHR\$(221)」をかく。

リスト 1

```
230 I=RND(24)+2:J=RND(15)+4  
250 IF SCR$(I,J)<>" " THEN 230  
270 LOCATE I,J:PRINT CHR$(221);:SE=SE-1
```

当然、リングでうめなければいけない空白は、一つへるわけだが  
ら、「SE=SE-1」。そして、ゲームを RUN させた直後はここを通  
らないようにしなければいけないので、

150 GOTO 300

を追加しよう。300 行からがゲームの始まりだ。



# 007-2

## どっちの方向なら 動けるか



まず、ペンペンとファイターフライのゲームの開始位置を決よう。  
230 行で「J=RND (15)+4」としてあるおかげで、新しくふえる  
カベは、上の4行には絶対ない。そこで、ペンペンは左上すみ、フ  
アイターフライは右上すみからスタートになるんだ。(この面での空  
白はSEだから、残りの空白を表わす変数SDにSEを入れよう。)

300 SD=SE POSITION 4,24,32

310 POSITION 2,216,32

さて、まずペンペンを動かそう。当然コントローラーの「STICK」  
関数を使うね。

400 V=STICK (0):N=4

ところが問題は、「STICK」の上下左右と「DEF MOVE」の上下  
左右の数値が違うんだ。そこで、

```
410 IF V=8 THEN V=1:GOTO 500
420 IF V=1 THEN V=3:GOTO 500
430 IF V=4 THEN V=5:GOTO 500
440 IF V=2 THEN V=7:GOTO 500
450 V=0:GOTO 510
```

ところが、これだけではまだ移動させてはいけないよ。移動予定  
先にカベがあるかどうかを調べなきゃいけないからだ。これは、フ  
アイターフライも同じ条件だから、一つのサブルーチンで処理しよ  
う。

500 N=4:GOSUB 6000

```
510 IF F=0 OR V=0 THEN
GOSUB 5000
```

6000 行で行けるか行けないかのマーク(フラグといったね。)を 0

●  
リス  
スト  
3

```
600 X=XPOS(4)/8-2
610 Y=YPOS(4)/8-3
620 FOR I=X TO X+1:FOR J=Y TO Y+1
630 IF SCR$(I,J)=" " THEN LOCATE I,J:PRINT CHR$(21
5);:SD=SD-1:SC=SC+1
640 NEXT: NEXT
650 IF SD<1 THEN 2000
```

にセット。X, Y はスプライト用の座標 (0~255) をキャラクター用の座標 (0~27, 0~23) に直したものだ。

6050 行から 6080 行は、行こうとする先にカベがあるかどうかを調べているんだ。

ペンペンに限っては、新しくできたカベ「CHR\$(221)」は通れなかったから、6070 行で調べている。

さて、サブルーチンから帰って、その方向に行けるなら (F=0), 「MOVE」のサブルーチンへ行って見て、だめなら (F=1), そのまま次へ行く。

●  
リス  
スト  
4


```
700 N=2:V=3+SGN(YPOS(4)-YPOS(2))*2:IF V<>3 THEN GO
SUB 6000:IF F=0 THEN 750
710 V=5+SGN(XPOS(2)-XPOS(4))*2:IF V<>5 THEN GOSUB
6000:IF F=0 THEN 750
720 GOTO 760
750 IF RND(100)<80 THEN 800
760 V=RND(4)*2+1
770 GOSUB 6000:IF F=1 THEN V=0
800 GOSUB 5000
```

この 600~640 行は 6050~6080 と同じような構文になってるね。ペ

ンペンのいるキャラクターサイズ2×2の下に、リンゴがなければ  
リンゴを書いて、その分SDをへらしているわけだ。そして、SDが  
0になれば、全面をリンゴでうめたことになるから、1面終わりの  
2000行へ飛ぶんだ。この650行を、

**650 GOTO 400**

とすると、とりあえずペンペンが動き回り、リンゴがふえていくの  
が見えるよね。





# 007-3

## 敵は おりこうさん



こんど  
今度はファイターフライを動かそう。

まず、考え方としては、乱数で1, 3, 5, 7を出すというのが  
ある。しかしそれでは、あまりにもファイターフライがバカすぎる。  
やっぱり、ファイターフライにはもう少し、りこうになってほしい。  
そこで、

●  
リス  
スト  
5

```
850 IF XPOS(2)<XPOS(4)+11 AND XPOS(2)>XPOS(4)-11 A
ND YPOS(2)<YPOS(4)+11 AND YPOS(2)>YPOS(4)-11 THEN
3000
```

700行と710行でペンペンのいる方向が、どっちかを決めている。  
このプログラムではY座標方向を優先しているが、700行と710行  
を入れかえれば、X方向が先になるのはわかるね。

ただ、ファイターフライには、少しおバカさんになってもらうこ  
とにする。750行で乱数によって、わざとでたらめな動きをしてもら  
うのだ。750行の、

IF RND(100)<80

の「80」を小さくすればするほど、乱数でしか動かない、おバカさ  
んのファイターフライになってしまう。といっても、これを「100」  
にすると完全なおりこうさんになるかというと、そうでもない。

ペンペンとファイターフライがスタート位置にいて、間に  
カベがあったらどうだろう。ファイターフライはペンペンに近よっ  
てくるが、カベの所まできると、もう動けなくなる。この程度のプ  
ログラムでは、カベを回りこんでくる、というのは、なかなかむず  
かしいからなんだ。だから少しは、乱数で動く可能性を残した方が  
いいんじゃないかな。

700行と710行ではメモリーの節約のため、6000行と6010行で使ったと同じ「SGN」関数をうまく利用しているね。

これで、ペンペンとファイターフライが自由に動き回れるようになった。いよいよ、ペンペンがファイターフライにつかまったかどうかを調べなきゃいけない。

5000行で決めているように、ペンペンもファイターフライも一回の「MOVE」の命令で8ドット動くんだから、XPOS(4), YPOS(4)のそれぞれ前後±11ドット以内に、ファイターフライが来たときを、つかまったとみなそう。

つまり、GAME OVER の処理は3000行で行なうわけだ。

●  
リス  
ト  
6

```
6000 F=0: X=XPOS(N)/8-2: IF V=3 OR V=7 THEN X=X+SGN(5-V)
```

```
6010 Y=YPOS(N)/8-3: IF V=1 OR V=5 THEN Y=Y+SGN(V-3)
```

```
6050 FOR I=X TO X+1: FOR J=Y TO Y+1
```

```
6060 IF SCR$(I,J)>CHR$(221) AND SCR$(I,J)<CHR$(232) THEN F=1
```

```
6070 IF N=4 AND SCR$(I,J)=CHR$(221) THEN F=1
```

```
6080 NEXT: NEXT: RETURN
```

# 007-4

## 最後の仕上げは こうやって



さて、得点<sup>とくてん</sup>はどうしよう？ リンゴ<sup>か</sup>1つを書くごとに、プラス1点<sup>てん</sup>としよう。とすると当然<sup>とうぜん</sup>、リンゴ<sup>か</sup>を書いて<sup>き</sup>いた630行<sup>ぎょう</sup>の一番最後<sup>ばんさいご</sup>に、

**SC=SC+1**

としなけりやだめだ。その得点<sup>とくてん</sup>を表示<sup>ひょうじ</sup>するために、

**950 GOSUB 7000**

としよう。(7000行<sup>ぎょう</sup>がスコア表示<sup>ひょうじ</sup>のサブルーチン<sup>こ</sup>だ。)

また、全部<sup>ぜんぶ</sup>の空白<sup>くうはく</sup>をリンゴ<sup>か</sup>でうめると、カベを1個<sup>こ</sup>ふやしてははじめからやるわけだ。しかし、スコア (SC) を0<sup>もど</sup>に戻<sup>もど</sup>してはまずいから、

**2000 GOSUB 7000**

**2100 GOTO 2000**

としなきゃあいけない。また得点<sup>とくてん</sup>を出<sup>だ</sup>して、得点表示<sup>とくてんひょうじ</sup>のサブルーチン<sup>こ</sup>7000行<sup>ぎょう</sup>と、一面クリアーのお祝い<sup>めん</sup>の言葉<sup>ことば</sup>2050行<sup>ぎょう</sup>、次の面<sup>つぎめん</sup>をはじめるとき<sup>とき</sup>、その言葉<sup>ことば</sup>を消<sup>け</sup>すための、280行<sup>ぎょう</sup>を追加<sup>ついか</sup>し、3000行<sup>ぎょう</sup>からの「END」処理<sup>しり</sup>をつけたすと、右<sup>みぎ</sup>のようになる。

このプログラムでは、まだまだメモリーにあまりがあるね。ジャンプ<sup>へん</sup>編<sup>へん</sup>で勉強<sup>べんきょう</sup>する言葉<sup>ことば</sup> (PLAY) をつけたり、ファイターフライを二匹<sup>ひき</sup>にしたりと、改造<sup>かいぞう</sup>する余地<sup>よち</sup>はいっぱいあるんだ。ぜひと、ガンバロー!

さて、いよいよジャンプ<sup>へん</sup>編<sup>へん</sup>では、音楽<sup>おんがく</sup>や色<sup>いろ</sup>に挑戦<sup>ちようせん</sup>してみようじゃないか。ゲームも、Tiny ROLL PLAYING GAMEだ!



```
100 SPRITE ON:CLS:VIEW
110 SC=0:SE=406
150 GOTO 300
200 FOR I=1 TO 27:FOR J=1 TO 19:IF SCR$(I,J)=CHR$(
215) THEN LOCATE I,J:PRINT " ";
210 NEXT:NEXT
230 I=RND(24)+2:J=RND(15)+4
250 IF SCR$(I,J)<>" " THEN 230
270 LOCATE I,J:PRINT CHR$(221);:SE=SE-1
280 LOCATE 10,23:PRINT " ";
300 SD=SE:POSITION 4,24,32
310 POSITION 2,216,32
400 V=STICK(0):N=4
410 IF V=8 THEN V=1:GOTO 500
420 IF V=1 THEN V=3:GOTO 500
430 IF V=4 THEN V=5:GOTO 500
440 IF V=2 THEN V=7:GOTO 500
450 V=0:GOTO 510
500 GOSUB 6000
510 IF F=0 OR V=0 THEN GOSUB 5000
600 X=XPOS(4)/8-2
610 Y=YPOS(4)/8-3
620 FOR I=X TO X+1:FOR J=Y TO Y+1
630 IF SCR$(I,J)=" " THEN LOCATE I,J:PRINT CHR$(21
5);:SD=SD-1:SC=SC+1
640 NEXT:NEXT
650 IF SD<1 THEN 2000
700 N=2:V=3+SGN(YPOS(4)-YPOS(2))*2:IF V<>3 THEN GO
SUB 6000:IF F=0 THEN 750
710 V=5+SGN(XPOS(2)-XPOS(4))*2:IF V<>5 THEN GOSUB
6000:IF F=0 THEN 750
```

```

720 GOTO 760
750 IF RND(100)<80 THEN 800
760 V=RND(4)*2+1
770 GOSUB 6000:IF F=1 THEN V=0
800 GOSUB 5000
850 IF XPOS(2)<XPOS(4)+11 AND XPOS(2)>XPOS(4)-11 AND YPOS(2)<YPOS(4)+11 AND YPOS(2)>YPOS(4)-11 THEN 3000
950 GOSUB 7000
999 GOTO 400
2000 GOSUB 7000
2050 LOCATE 10,23:PRINT "GOOD !";
2100 GOTO 200
3000 BEEP:PAUSE 230:BEEP:CLS:SPRITE OFF
3010 LOCATE 7,10:PRINT "SCORE =";SC
3020 LOCATE 5,13:PRINT "HIT ANY KEY !"
3030 PAUSE
3040 RUN
3100 END
5000 DEF MOVE(N)=SPRITE(N,V,1,4):MOVE N:RETURN
6000 F=0:X=XPOS(N)/8-2:IF V=3 OR V=7 THEN X=X+SGN(5-V)
6010 Y=YPOS(N)/8-3:IF V=1 OR V=5 THEN Y=Y+SGN(V-3)
6050 FOR I=X TO X+1:FOR J=Y TO Y+1
6060 IF SCR$(I,J)>CHR$(221) AND SCR$(I,J)<CHR$(232) THEN F=1
6070 IF N=4 AND SCR$(I,J)=CHR$(221) THEN F=1
6080 NEXT:NEXT:RETURN
7000 LOCATE 8,21:PRINT "SCORE =";SC:RETURN

```

わが地球防衛軍スタッフの連日のわたるファミコンとの闘いで、  
以下の事が判明した。

## CALL

① Ver. 2 でも後期に印刷されたマニュアルには書かれているんだが、「CALL」という命令がかくされていた。これはマシン語プログラムを動かす命令である。使い方は、ジャンプ編でくわしくのべよう。

② 他にも「SPC」と「TAB」という命令が登録されていた。ただし、これは名前が登録されているだけで使えない。Ver. 4 以降にでも使うつもりかもしれない。

## ③ バグ情報-1

Ver. 3 で作ったプログラムをテープへSAVEしたものを Ver. 2 でLOADすると、プログラムがメチャメチャになってしまう、というのは、ホップ編や、ゲームプログラム集 001 でもすでに説明したが、やっと原因がわかったので、説明しよう。  
コンピューターが数字をメモリーへ入れる時、たとえば

LOCATE 0, 1

## TAPE ばけ

とかいうものの「0」や「1」は、まず最初の場所に、「次は10進法の数字だよ」というしるし(コード)を書いて、そのあと2バイトを使ってその数字を入れている。(つまり1つの数字を記憶するのに3バイト必要だ)ところが、Ver. 3 では、何を考えたか、1ケタの数字を記憶する時に限って、1バイトしか使わないように変更してしまったんだ。たしかに、メモリー数の少ないファミコンでは、メモリーの節約は非常にありがたいんだけど、この方式で書かれたテープを、Ver. 2 で読むと、大変なことになる。Ver. 2 では、数字をあらわすのに必ず3バイト使っているから、プログラムを見に行く順番が狂ってしまうのだ。

命令 [ ] すうじ [ ] 命令 Ver. 2

命令 [ ] 1ケタの数字 [ ] 命令 Ver. 3

Ver. 3 をもっている人はためしてほしい。

10 LOCATE 10, 12



10 LOCATE 0, 2

この2つのプログラムで残りバイト数(PRINT FREでよかったね)が変る事がわかるだろう。

対策としては

10 LOCATE 0, 1

のような1ケタの数字の前に、必ず

10 LOCATE &H0, &H1

のように「&H」という文字を入れておけば大丈夫だ。この「&H」というのは16進数という数のかぞえ方を意味するやっかいなものだ。(ジャンプ編でくわしく説明する)

ただし、どんなことがあっても、2ケタ以上の数字の前にはつけない事(10→&H10のように)、数字の中身が変ってしまうからだ。

10 LOCATE 10, 12

10 LOCATE &H0, &H2

この二つでは残りバイトが一緒だというのは調べればすぐわかる。

## COLOR文 の位置ずれ

### ④ バグ情報-2

以下のプログラムを実行してみよう。

10 CLS

20 LOCATE 8, 8

30 PRINT "TEST1"

40 LOCATE 8, 9

50 COLOR 8, 9, 3

60 PRINT "TEST2"

「LOCATE」文で、X座標を「8」に指定しているにもかかわらず、60行のPRINT文の文字列は、ずれて表示されている。つまり、COLOR命令を実行するとカーソルの位置がずれてしまうのだ。

10 LOCATE 8, 8

20 A=POS(0)

30 COLOR 8, 8, 3

40 B=POS(0)

50 CLS

60 PRINT A, B

このプログラムでわかるだろう。このバグはVer.3では解消され

## カタカナの化け

ているが、Ver. 2ではLOCATE<sup>めいれい</sup>命令をともなったCOLOR文は必<sup>かなら</sup>ず、PRINT文<sup>ぶん</sup>のうしろに置く<sup>お</sup>ことが必要<sup>ひつよう</sup>だ。

### ⑤ バグ情報—3

```
1 Ø READ A$
2 Ø PRINT A$
3 Ø DATA ヤ
```

として「RUN」してみよう。ちゃんと「ヤ」と表示<sup>ひょうじ</sup>されるはずだ。ここで「LIST」をとってみると、

```
1 Ø READ A$
2 Ø PRINT A$
3 Ø DATA READ
```

となっている。

これは、「LIST」の処理<sup>しり</sup>ルーチンにバグがあるため、「DATA」のうしろにカタカナがあると、ものによっては中間言語<sup>ちゅうかんげんご</sup>（ファミリーベーシック・ジャンプ<sup>へん</sup>編<sup>み</sup>を見て！）と間違<sup>かんが</sup>いしてしまうせいだ。

```
3 Ø DATA "ヤ"
```

のようにすれば問題<sup>もんだい</sup>はないが、普通<sup>ふつ</sup>ほとんどのパソコンも、DATA文<sup>もん</sup>の文字列<sup>もじれつ</sup>は「"」で囲<sup>かこ</sup>まれていなくてもよいという約束<sup>やくそく</sup>なので困<sup>こま</sup>ってしまう。「ヤ」以外<sup>いがい</sup>にもこういう変化<sup>へんか</sup>をするカタカナはいくつもある。探<sup>さが</sup>してみよう。また、変化<sup>へんか</sup>する時<sup>とき</sup>としない時<sup>とき</sup>がある。なお、このバグはVer. 3でもとれていない。

## カナ文字モードの RETURN KEY

### ⑥ バグ情報—4

どのコンピューターも、「英数字モード<sup>えいすうじ</sup>」と「カナ文字モード<sup>もじ</sup>」のどちらの状態<sup>じょうたい</sup>であっても、RETURNキーの機能<sup>きのう</sup>は同じ<sup>おな</sup>のはずだが、我々スタッフ<sup>われわれ</sup>がEDIT（行<sup>ぎょう</sup>の中身<sup>なかつみ</sup>を修正<sup>しゆせい</sup>する）中に、「カナ文字モード<sup>もじ</sup>」の状態<sup>じょうたい</sup>でRETURNキーを押したあと「LIST」を取<sup>と</sup>ると、修正<sup>しゆせい</sup>されていない（RETURNキーの機能<sup>きのう</sup>が働<sup>はたら</sup>いてない）という現象<sup>げんしょう</sup>が何回<sup>なんかい</sup>もあった。あとでのべる「熱暴走<sup>ねつぼうそう</sup>」のせいかもしれないが、重要な修正<sup>じゆうよう しゆせい</sup>は「英数字モード<sup>えいすうじ</sup>」にしたあとRETURNキーを押した方<sup>ほう</sup>が安心<sup>あんしん</sup>だ。

### ⑦ バグ情報—5

次のプログラム<sup>つぎ</sup>を入力<sup>にゅうりよく</sup>してほしい。

```
1 Ø Ø READ A$, B
```

DATA  
READ

1 1 0 PRINT A\$, B

2 0 0 DATA "C"

2 1 0 DATA 1

こうすると「A\$」にはちゃんと「C」という文字が入っているけれど、数値変数「B」にはDATA文にある「1」ではなくて「0」が入っている。

ところが、この200行と210行を

2 0 0 DATA "C", 1

または、

2 0 0 DATA C

2 1 0 DATA 1

のようにすると、「B」にはちゃんと「1」が入っている。

これは、おそらくDATA文の中で「"」のあとに何も無い時は、「,」の役割も一緒にしているせいだと思われる。

つまり、

2 0 0 DATA "C",

と書いたのと同じわけだ。

コンピューターは「C」という文字を読んだ(READ A\$)あと、その「C」のうしろに「,」があると思う。ところが「,」のうしろには何も無いので、とりあえず「0」を次のREADに代入してしまうというわけだ。

おもしろいのは「"」のうしろに「,」があって、他のデータが続くときは「"」は「,」の働きをしなくなっている。

このバグを防ぐには、

2 0 0 DATA "C", 1

のように「"」のうしろに他のデータをつけるか、「"」を使わないか、しかないが、バグ情報-3に書いたように、「"」をつけないとカタカナの文字列がLISTをとるとバケてしまう。DATAをREADする時、文字変数よりあとに数値変数を置くのも、プログラムの内容によっては不可能な時もある(RESTOREなどの時)。

「ソフト集 002」では、ダミーのデータを一個おいているが、困ったバグだ。

このバグは、Ver. 3ではなくなっているようだ。



## ⑧ バグ情報-6

これは、「バグ」ではないが、それ以上に

注意!

よう ちゆう い  
要 注 意 !

「<sup>ねつぼうそう</sup>熱暴走」という<sup>ことば</sup>言葉を知っているだろうか。古いパソコンではよくあった。<sup>ちようじかん</sup>長時間コンピューターの<sup>でんげん</sup>電源を入れているうちに、コンピューターが<sup>ねつ</sup>熱を持ち、その<sup>ねつ</sup>熱がコンピューターに<sup>えいきよう</sup>影響を与え、<sup>せいじよう</sup>正常な動作をしなくなることがある。

現在、わが<sup>げんざい</sup>スタッフは12台の<sup>だい</sup>ファミコンを持ち、<sup>も</sup>日夜<sup>にちや</sup>開発と<sup>かいはつ</sup>デバッグにはげんでいるのだが、その内の<sup>うち</sup>約半数が時々<sup>やくはんすう</sup>暴走してしまうのだ。

プログラムを<sup>にゆうりよくちゆう</sup>入力中に、突然、あの<sup>とつぜん</sup>恐怖の<sup>きようふ</sup>オープニング画面と<sup>がめん</sup>共に「ピコピコ」と音が出てきたり、カーソルが<sup>おと</sup>消えて<sup>き</sup>キーボードをたたいても何の<sup>なん</sup>反応もなくなってしまうのだ。

たしかに、<sup>てつや</sup>徹夜を<sup>つづ</sup>続けて何日間も<sup>なんにちかん</sup>電源を入れっぱなしの、かなりハードな<sup>つか</sup>使い方をしているのは<sup>かた</sup>事実だが、これではあまりにも<sup>じじつ</sup>スタッフ<sup>じ</sup>がかわいそう。

<sup>でんげん</sup>電源はこまめに切りましょう。また、<sup>こ</sup>コンセントにさし込んだ<sup>でん</sup>電源アダプターも、ファミコンを<sup>つか</sup>使わない時はぬいておきましょう。

それにしても、<sup>しょうしょうこしよりつ</sup>少々故障率が<sup>たか</sup>高すぎる<sup>き</sup>気もするのだが……。

## 作戦資料 102 ROMとRAM

プログラムを<sup>つく</sup>作ったあと、テープに<sup>でんげん</sup>SAVEしないと、<sup>き</sup>電源を切ったら全部消えちゃうのは、もうわかったよね。じゃあ、ゲームの<sup>ぜんぶ</sup>カセットや、ファミリーベーシックの<sup>でんげん</sup>カセットは、なぜ<sup>い</sup>電源を入れるたびに、<sup>おな</sup>同じ働きをするんだろう? ゲームカセットの中にもゲームプログラムが入っているはずだし?

<sup>かんが</sup>そう考えた君は、<sup>きみ</sup>するどい!<sup>そうとう</sup>相当にするどい!

パソコンのMSXのゲームも、テープで<sup>はつばい</sup>発売されているのもあれば、<sup>はつばい</sup>カセットで<sup>ちが</sup>発売されているものもある。どこがどう<sup>ちが</sup>違うんだろう?

<sup>きみ</sup>君の考えた通り、<sup>かんが</sup>カセットのゲームも、<sup>とお</sup>テープのゲームも、<sup>ほんとう</sup>本当

はベーシック自体も、一つのプログラムなんだ。じゃあ、なぜ消えるプログラムと、消えないプログラムがあるんだろう？

それは、プログラムのせいではなく、プログラムを書きこんでいる「もの」が違うせいなんだ。

例えば、クーラーなんかにも、マイコン内蔵型というのが発売されてるよね。まさか、クーラーに入ったマイコンを使ってゲームをする人は、いないだろ。「ドンキー・コング」のカセットで「ゼビウス」はできないだろう。印刷された「本」と、まっ白な「ノート」の違いは、わかるだろう。1冊の本の中身は、いつまでたっても変わらない。それに比べて、ノートにはいろんな物をかいたり消したりして、別の物を書ける。

つまり、本やクーラーのマイコンのように、使い道が完全に決まっているものは、最初から印刷した方がいいだろう。白紙の本を買ってきて、それにテープか何かから、内容を写したあとで読む、なんて、だれもやんないよね。

コンピューターのプログラムを記憶しておく「容器」には「ROM」と「RAM」という二種類あるんだ。

Read Only Memory

読みだし専用メモリー

Random Access Memory

自由に使えるメモリー

のそれぞれ頭文字をとった略称なんだ。

「ROM」にプログラムを入れる（焼き込む、ともいう）には、特別な機械（といっても、最近はやすもので4～5万円であるけど）が必要だ。そのかわり、一回焼き込むと電源を切っても中身は保存されている。

普通のコンピューターでは「ROM」からプログラムを読み出すことはできても、書き込むことはできない。つまり、読み出し専用メモリー、というわけだ。

「RAM」の方は、コンピューターで自由に書いたり読んだりできるけど、電源が切れると、中身も消えちゃう。電源を切ると、全ページに消しゴムを使うのと同じだから、電源を入れたときは、いつも白紙のノートとして使えるんだ。

だから、「RAM」は、コンピューターのプログラムのように、そのつどに別の仕事をさせるために、中身がいつも変わってないといけないものを記憶するときに使うんだ。

ゲームカセットやベーンシックは、この「ROM」の中身にプログラムを入れていたので、電源を切っても中身が消えなかったわけなんだ。

ただ、ゲームカセットでは「ROM」だけしか使わないかという、と、そうでもないんだ。得点や、主人公が今どこにいるか、というように刻々と変化するデーターは、いつも、書き込み、読み出しをしなければいけないだろう。

#### ファミコンのRAM

ファミコン内部には、そういうデーターを入れる「RAM」(ワークエリア)が、ちゃんと入っている。

せっかく出した「ハイスコア」が、電源を切るとなくなっちゃうのは、やはり、得点を書きこんでいるのが「RAM」だから、というわけだ。

また、「ROM」に一回書き込むと、絶対中身が消えないか、という、とそうでもないんだ。「ROM」の材料によっては、家庭用の蛍光灯に長い時間くっつけておくと消えてしまうものや、特殊な機械で、3~4回は書きかえることができるものもある。

でも、お願いだから、実験はしちゃだめだよ！もう一台、買わなきゃならないよ……。

## 作戦資料 103 コントロール キーの使い方

Ctr-

ホップ編でも書いたけど、このコントロールキー、なくても別に、かまわない。でも、作戦005で使った「Ctr-D」なんて、すごく便利だ。

そこで、ここでは、コントロールキーの使い方をくわしく書いてみよう。

「Ctr-D」というのは、「CTR」キーを押しながら、「D」のキーを押すことだったね。

この「Ctr-」で使えるキーは次の14個だ。

A, C, D, E, G, H, J, K, L, M, R, V, W, Z



普通のパソコンだと「カナ文字」モードだと使えないことが多いけど、ファミコンでは大丈夫なんだ。

まず、キーボード上の特殊キー（ホップ編参照）と同じ働きをするのが8個だ。

- C t r-H 「DEL」キー
- C t r-J カーソルキーの「▼」
- C t r-K 「HOME」キー
- C t r-L 「SHIFT」キーを押しながら「CLR・HOME」キー
- C t r-M 「RETURN」キー
- C t r-R 「INS」キー
- C t r-V カナ文字モードにする
- C t r-W 英数字モードにする

このうち「Ctrl-V」と「Ctrl-W」は「カナ」キーと同じだけど、「カナ」キーは、一回押すごとに、カナ文字モードと英数字モードが切りかわったのに対し、それぞれの役割を分けているんだ。

さて、何かのプログラムを打ち込んでみよう。どこか文字の書かれている所で、

#### C t r-E

を押してみよう。その行のカーソルからあとが、全部消えたのがわかるかな。

今度は別の文字の上へカーソルを持っていて、

#### C t r-Z

と押してみる。これは、その行だけじゃなく、カーソルの下の行も全部一変に消えてしまった。

ただし、この「Ctrl-E」、「Ctrl-Z」ともに、表示されているのを消しただけで、リターンキーで押さない限り、プログラムは消えてはいないんだ。場所を限定した「CLR」キーだと思えばいいかな。

#### C t r-G

とすると、「ビー」という音がしたね。ベーシックの命令の「BEEP」と同じ働きだね。

#### C t r-D

これは、作戦005で何回も使ったね。スプライトの所で、すべての

ひようじ　しよ　き　せつてい　でんげん　い　おな　じようたい　やくめ  
表示を初期設定（電源を入れたときと同じ状態）にする役目だ。

## CGEN 2

### SPRITE OFF

Ctr-A (INSモード)の解除

カラーパレットをスプライト・キャラクター面とともに、キ  
ャクター用のパレットコード1にする。

の4つの働きを一度にしてしまうのだ。

### Ctr-A

「INS」キーは、一回押すと空白を一個、カーソルの手前にわりこ  
ます働きを持っていた。ところが、この「Ctr-A」は少し違うん  
だ。

### TEST

と文字を入力して、カーソルを「S」の上に持っていこう。そこ  
で、

### Ctr-A

とする。何も変化しない、ようにみえる。

ここで、たとえば「D」のキーを押してみよう。

### TEDST

となって、カーソルは「S」の上だね。いろんなキーを押してみよ  
う。次々とその文字が「S」の前にわり込んでくる。

この状態を「INS」モード、つまりこれが「わり込みモード」な  
んだ。ここでもう一回、

### Ctr-A

を押すと、今度はこの「INS」モードが解除されている。「カナ」キ  
ーと同じように、一回押すごとに、ON、OFFと変化するわけなん  
だ。

### Ctr-D

でも、この「INS」モードを解除できるよ。

最後にコントロールコード Ctr-C は、プログラム入力中に操  
作すると、その行は登録されません。（プログラム入力中にSTOP  
キーを押すのと同じ働きです。）



# キャラクターテーブルA

0	1	4	5	8	9	12	13	16	17	20	21	24	25
2	3	6	7	10	11	14	15	18	19	22	23	26	27
28	29	32	33	36	37	40	41	44	45	48	49	52	53
30	31	34	35	38	39	42	43	46	47	50	51	54	55
56	57	60	61	64	65	68	69	72	73	76	77	80	81
58	59	62	63	66	67	70	71	74	75	78	79	82	83
84	85	88	89	92	93	96	97	100	101	104	105	108	109
86	87	90	91	94	95	98	99	102	103	106	107	110	111
112	113	116	117	120	121	124	125	128	129	132	133	136	137
112	113	116	117	120	121	124	125	128	129	132	133	136	137
114	115	118	119	122	123	126	127	130	131	134	135	138	139
140	141	144	145	148	149	152	153	156	157	160	161	164	165
140	141	144	145	148	149	152	153	156	157	160	161	164	165
142	143	146	147	150	151	154	155	158	159	162	163	166	167
168	169	172	173	176	177	180	181	184	185	188	189	192	193
170	171	174	175	178	179	182	183	186	187	190	191	194	195
196	197	200	201	204	205	208	209	210	211	212	213		
196	197	200	201	204	205	208	209	210	211	212	213		
198	199	202	203	206	207								
198	199	202	203	206	207								

これらのキャラクターは、主にアニメキャラクターとして使用します。  
各キャラクターの四隅に表記している数字は、DEF SPRITE文のCHR\$(n)で使用する10進数です。



# キャラクターテーブルB

	0	1	2	3	4	5	6	7
A								
B								
C								
D								
E								
F								
G								
H								
I								
J								
K								
L								
M								

検印省略

NDC 548

---

## ファミリーベーシック ステップ

昭和60年6月24日 発行

定価 800円

著 者 地球防衛軍

発行者 小川 茂 男

発行所 (株)誠文堂新光社

東京都千代田区神田錦町1-5-5

郵便番号 101

電話 東京 (292) 1211

振替口座 東京 7-6294

印刷・広研印刷株式会社

製本・岡嶋製本工業

© 地球防衛軍

Printed in Japan

(本誌掲載記事の無断転用を禁じます)

万一落丁・乱丁の場合はお取替えいたします

ISBN4-416-18527-8 C2055

---

本社発行 子供の科学／天文ガイド／MJ無線と実験／初歩のラジオ／農耕と園芸  
の雑誌 商店界／アイデア／ブレーン／ザ・コピーライターズ／月刊 芽  
ガーデンライフ／愛犬の友／フローリスト／囲碁／DEVICE file／ポ  
ートフォリオ





倉敷市立中央圖書館	
倉敷市立図書館	
分類	79
除籍	
20.02/20	
60.0.25	

## ファミコン大作戦

ファミコンゲーム大公開の第1弾

## ファミコンソフト集 001

全14本のゲームソフトが大集合

☆FLYFLY ☆アステロイド ☆FLYFALL  
☆コマンダー ☆いも虫ごろごろ ☆カメレース  
☆チックタックBOMB ☆マリオでジャンケン  
☆MARIOKICK ☆ライフゲーム ☆ラリー  
☆ガールハント ☆MOO ☆REFRECTION


地球防衛軍 編著 B5変形108頁 定価800円





ファミコン大作戦

ファミリーベーシック™ ステップ



# ファミコン大作戦

誠文堂新光社

ISBN4-416-18527-8 C-2055 ¥800E 定価 800 円

●ファミリーコンピュータは任天堂の商標です。



大作戦

ヒーロー

ヒーロー

誠文堂

